
Preface

As we complete the final edits and our printing deadline looms, we•re excited and 
grateful to be involved with JavaFX. In February 2009, JavaFX reached the 
100,000,000th download of the JavaFX runtime.1 The ranks of JavaFX developers will 
undoubtedly grow as more developers see the flexibility and power of JavaFX. This 
synergy, we believe, will fuel continued development of the language and enhance-
ments to the runtime environment.

This book is designed to get you up to speed quickly with JavaFX. JavaFX is a script-
ing language. It•s not Java, but it•s built on top of the Java runtime. You don•t need 
experience with Java to succeed with JavaFX. Indeed, JavaFX•s declarative syntax 
makes life easier if you don•t think like a developer. Instead, JavaFX encourages you to 
think like a designer. 

What does it mean to •think like a designerŽ? Basically, it means to visualize the struc-
ture of your application or widget and compose your scene out of simple shapes and 
other building blocks. In JavaFX, you compose a scene by declaring objects.

Let•s take an example. Say you visualize a sky with the sun, the sea, and an island 
(think South Pacific). The sky is the background, reflecting the blues of a bright cloud-
less day (think of a linear gradient, going from •blue skyŽ to •azureŽ). The sun is a 
Circle, with a radial gradient consisting of yellows and oranges. The island is a qua-
dratic curve (think of a cone-shaped volcano-type island paradise filled with a gradi-
ent of rich browns and tropical greens). And there you have your scene, as shown in 
Figure 1 (in a black and white approximation). 2

Not only can you declare visual objects with JavaFX, but you can also declare anima-
tions. Animations give your objects life. Returning to our island paradise, visualize 
the beginning of the day. The colors are muted as the morning light slowly gives 
shape to an ethereal world. The sun rises and the island takes form. The sun continues 

1. Jonathan Schwartz•s Blog: JavaFX Hits 100,000,000 Milestone! February 13, 2009. URL: 
http://blogs.sun.com/jonathan/entry/javafx_hits_100_000_000

2. You•ll find widget Island Paradise with the other JavaFX examples on the authors• web site 
at http://www.asgteach.com/javafx .

higher in the sky and the whole world brightens. When the sun reaches its zenith, the 
colors of the sea and sky are saturated with •sea green,Ž •azure,Ž and •sky blue.Ž But, 
it•s a temporary brightness. The sun follows its path and eventually falls back into the 
sea in a glow of warm reds. The sky darkens. The sea fades. Eventually the island dis-
appears. Everything is black. You build these behaviors with animation and a power-
ful JavaFX language construct called binding. (Binding lets you declare dependencies 
among objects„when a variable changes, JavaFX automatically updates all objects 
bound to that variable.)

JavaFX animation lets you move objects along a path (the sun rises and sets in an arc) 
and fade objects in and out with timelines. Our •island paradiseŽ controls day and 
night with black and red rectangle •filters.Ž These filters color the scene as night, 
dawn, daytime, evening, dusk, and back to night, all cycling through an accelerated 
Circadian clock. Animation and binding make it all work.

If you•re a Java programmer, you will feel at home in JavaFX with packages and 
import statements, classes, things called public, and static type checking. (Don•t worry 
about these things if you•re not a Java programmer.) If you•re a JavaScript program-
mer, you will appreciate the value of static type checking coupled with a sophisticated 
type inference engine in JavaFX. (Type inference eases the burden of having to specify 
types everywhere.) But most importantly, we hope you•ll appreciate the simplicity of 
the JavaFX declarative style. For example, take this one-line JavaFX object •literal.Ž

Circle { centerX: 200 centerY: 40 radius: 25 fill: Color.YELLOW }

If you think the above describes a yellow circle, then you•re on your way! And, if you 
think perhaps it describes a yellow sun, that•s even better.

Figure 1.  Visualizing an island paradise

About the Audience

This book is aimed at developers with some previous programming experience (in 
any language). We don•t assume you know Java and we assume you•ve never written 
a JavaFX script before. (The term script in this book refers to both the programs you 
write and the individual files that contain these programs.)

We hope to show you how to use JavaFX effectively. There is a diversity to JavaFX: 
you can use it to build games, create effective web-service-based widgets, or build 
snazzy front-ends to your desktop applications. You can use Swing-based compo-
nents, •nativeŽ JavaFX components, or roll your own. You can collaborate with 
designers and import images and other assets to incorporate into your scene graph. 
Our aim is to expose some of this diversity so that you can forge ahead with your own 
successful JavaFX projects.

How to Use This Book

Chapter 1 gets you started with JavaFX. We show you how to download JavaFX and 
begin building projects with the NetBeans IDE. (We use NetBeans to build our exam-
ples, but you can also use Eclipse.)

Chapter 2 gives you a broad overview of JavaFX. It takes you through an example (a 
Guitar Tuner), pointing out how things are done with JavaFX. If you want to get a 
•feelŽ for the language, this chapter introduces you to many trademark JavaFX fea-
tures.

Chapter 3 through Chapter 5 are •reference-orientedŽ chapters. Chapter 3 describes 
the JavaFX language, Chapter 4 describes graphical objects, and Chapter 5 discusses 
user interface components. These chapters are organized with small examples to help 
you find information quickly (how do I bind an object or generate a sequence with a 
for loop?). The language chapter covers everything from JavaFX built-in types to 
mixin inheritance. Graphical objects are the basic JavaFX shapes you use to build 
scene graphs and layout objects (islands in the sun, for example). The components 
chapter shows you the JavaFX Swing components and the JavaFX •nativeŽ UI compo-
nents. We also show you how to build custom UI components in a more advanced 
section.

Chapter 6 shows you how to design and structure a JavaFX application. It introduces 
a building-block approach with a nod towards object oriented design principles.

Chapter 7 is all about JavaFX animation and timelines. JavaFX animation is both pow-
erful and flexible. Transitions are •pre-packaged high-levelŽ animations that help 
build straightforward motions quickly, such as fade-ins and fade-outs, scaling, and 
moving.

Chapter 8 discusses viewing and manipulating images. One example shows you how 
to design an animated photo carousel.

Chapter 9 covers web services. JavaFX provides two important utility classes that 
make it easier to work with web services. An HttpRequest class handles asynchro-
nous web requests and a PullParser class simplifies processing the response data. We 
take you through several Flickr-based web service API calls.

Chapter 10 discusses the JavaFX mobile environment and explores the differences 
between desktop JavaFX and the JavaFX mobile runtime. We discuss guidelines for 
targeting mobile devices and how to make an application mobile-friendly.

About the Examples

You can download the source code for all book examples from the authors• web site at

http://www.asgteach.com/javafx

In addition, example applications are deployed so you can try them out.

Notational Conventions

We•ve applied a rather light hand with font conventions in an attempt to keep the 
page uncluttered. Here are the conventions we follow.

Element Font Example

JavaFX class Shape, Circle, Color

JavaFX property layoutBounds, opacity, height

JavaFX code def sunPath = Path {

    elements: sunElements

    stroke: Color.GRAY

}

URL http://javafx.com/

file name Main.fx , Carousel.fx

key combinations Ctrl+Space

NetBeans menu selections Properties  menu item

code within text The animation varies property opacity  from . . . 

code highlighting 

(to show modified or relevant 
portions)

def sunPath = Path {

elements: sunElements

    stroke: Color.GRAY

}

1

1 Getting Started with 
JavaFX

Welcome to JavaFX. This chapter tells you what JavaFX is, what it does, how to get it, 
and how to get started. After you finish this chapter, you should be able to start work-
ing with JavaFX right away.

What You Will Learn

€ What is JavaFX and why should I care?

€ What is in the JavaFX Bundle

€ Where to get JavaFX

€ Using JavaFX with the NetBeans IDE

1.1  What Is JavaFX?

JavaFX is a software technology that lets you create and deliver Rich Internet Applica-
tions (RIAs) with media and content across a wide variety of platforms and devices. 
The language was originally called F3 (Form Follows Function) and was developed 
primarily by Chris Oliver, now at Sun Microsystems. The name was changed to 
JavaFX in 2007.

On Java platforms, JavaFX is a compiled, statically typed, declarative scripting lan-
guage. The language offers automatic data binding, triggers, animation, and an 
expression syntax where code blocks yield values. Sequences, function types, and 
inferred types make JavaFX a concise scripting language. Developers can use object-
oriented JavaFX features to simplify complexity and handle errors with Java-like 
exceptions. JavaFX also lets you easily access the complete Java API, which includes 
an enormous number of third-party Java libraries.

Figure 1.1 is the big picture for JavaFX. This block diagram shows the various pieces 
you can leverage when creating JavaFX applications. As the diagram shows, there are 
extensions for three environments: Desktop, Mobile and TV. All three environments 
support the common API (labeled Common Elements in Figure 1.1). As of this writ-
ing, the runtime for the JavaFX TV environment does not yet exist. There is a runtime 

for the JavaFX mobile environment, which supports the common API only. (We dis-
cuss the JavaFX mobile environment in Chapter 10.)

On the desktop, JavaFX runs on Windows XP, Vista, and Mac OS. Linux support is 
forthcoming (as of this writing). JavaFX is also capable of running on Android, Win-
dows Mobile, and other mobile operating systems.

1.2  The JavaFX Bundle

The JavaFX platform release currently includes three major components. 

JavaFX SDK

The JavaFX SDK (Software Development Kit) includes the JavaFX compiler, runtime 
tools, graphics, media, and web services. It also includes libraries that let you create 
RIAs (Rich Internet Applications) for desktop, browser, and mobile platforms.

Figure 1.1 JavaFX Platform

Application Framework

Common Elements

JavaFX Runtime

JavaFX
Desktop
Runtime

Desktop
Extensions

JavaFX
Mobile

Runtime

Mobile
Extensions

JavaFX
TV

Runtime

TV
Extensions

Java Virtual Machine (Java Plug In)

NetBeans IDE

NetBeans is a sophisticated IDE (Integrated Development Environment) that lets you 
build, preview, and debug JavaFX applications. The code editor supports JavaFX syn-
tax checking, code completion, hyperlinked documentation, and other developer-
friendly features. To decrease development time, the editor offers drag-and-drop from 
palettes of GUI controls, event handlers, transformations, effects, and animation. For 
Mobile applications, NetBeans also supports a Mobile emulator to simulate applica-
tions running on mobile devices.

Currently, JavaFX is a plug-in component for NetBeans. A community-supported 
plug-in for Eclipse IDE users is also available.

JavaFX Production Suite

The JavaFX Production Suite is a set of tools and plug-ins to help designers export 
JavaFX graphics from third-party applications (Adobe Illustrator and Photoshop). 
Using the JavaFX Graphics Viewer, you can preview how graphics will render when 
you deploy to desktop and mobile environments. An SVG (Scalable Vector Graphics) 
conversion tool lets you convert SVG files to JavaFX format.

1.3  Where to Get JavaFX

To access the JavaFX downloads, go to http://java.sun.com/javafx/downloads . You 
will see downloads for the following components.

€ JavaFX SDK

€ NetBeans for JavaFX

€ JavaFX Production Suite

Follow the instructions on the web site to download and install the component you 
want. You can download and install NetBeans with JavaFX together or you can install 
the JavaFX plug-in separately if you already have NetBeans installed.

Here are some other valuable links for JavaFX.

€ http://java.sun.com/javafx „Main site for JavaFX

€ http://javafx.com „Samples and demos for JavaFX

€ http://java.sun.com/javafx/ num/docs/api „JavaFX documentation (version num)

€ http://www.netbeans.org „NetBeans site

€ https://openjfx-compiler.dev.java.net „OpenJFX Compiler Project

Once you have NetBeans and JavaFX installed, you are ready to try out JavaFX. 

Create a NetBeans Project

In this section, we show you how to build a NetBeans project for a JavaFX application, 
compile a short program, and run it. 

Launch NetBeans and choose File>New Project  as shown in Figure 1.2. 

You will see the New Project dialog, as shown in Figure 1.3.

Figure 1.2 NetBeans Create New Project

Under Categories, select JavaFX and click Next  as shown. Another dialog will appear 
that lets you chose a Project name, location, and project configuration, as shown in 
Figure 1.4.

Figure 1.3 NetBeans New Project Dialog

To finish specifying your JavaFX project, perform these steps:

1. Specify CoolApp  for the Project Name. 

2. Accept the default for Project Location  or click the Browse button to change the 
location of this project.

3. Make sure the checkboxes for Set as Main Project and Create Main File  are 
checked.

4. Click Finish .

Netbeans creates the CoolApp application and brings you up into the source editor as 
shown in Figure 1.5.

Figure 1.4 Choose Project Name and Location

Edit JavaFX Source Code

You are now ready to edit the source code. Move the scrollbar in the editor window to 
the bottom, as shown in Figure 1.6.

Figure 1.5 CoolApp Application

Perform these steps to edit the source code. (The line numbers in your edit window 
may not exactly match. We reference the line numbers in Figure 1.6.)

1. Show line numbers. (Place the mouse in the left column of the editor window, right 
click, and select Show Line Numbers  from the context menu.)

2. Change "Application title"  to "Cool App"  (line number 19 in Figure 1.6).

3. In line number 25, change 16 to 24. The line should now read size : 24 .

4. Change "Application content"  to "JavaFX is Cool"  (line number 28).

Compile and Run 

Click the green chevron on the tool bar to Run the Main Project, as shown in 
Figure 1.6. This will compile and run your application. You should see the application 
in a window on your screen, as shown in Figure 1.7.

Figure 1.6 Edit Source Code

Execution Models

Netbeans lets you run your JavaFX programs in several different environments. Your 
choices are as follows.

€ Standard Execution (default)

€ Java Web Start

€ Run in Browser

€ Run in Mobile Emulator

Standard Execution (the default) is for desktop applications. The Mobile Emulator 
simulates the Mobile environment. You can also run your application in a browser or 
use Java Web Start, a tool for deploying desktop applications with the Java Network 
Launching Protocol (JNLP).

To choose a deployment option, right-click the CoolApp project in the Project view, 
and choose Properties from the drop-down list, as shown in Figure 1.8.

Figure 1.7 Run CoolApp Application

NetBeans displays the Project Properties dialog. Select Run under Categories as 
shown in Figure 1.9. 

NetBeans now displays a new Project Properties dialog, as shown in Figure 1.10. 

Figure 1.8 Choose Properties

Figure 1.9 Choose Run under Categories (Project Properties)

The Standard Execution is pre-selected for you, but you can change to any of the other 
execution models. After you make your selection, click OK . When you run your appli-
cation with the green chevron on the tool bar as before (see Figure 1.6 on page 8), your 
application will run in the environment you selected.

NetBeans Tip

The Project Properties dialog lets you specify arguments to your program when you run it. 
You can also specify JVM (Java Virtual Machine) arguments.

Figure 1.10 Run CoolApp Application

2 A Taste of JavaFX

As the preface hints, JavaFX has a combination of features that makes it unique. This 
chapter gives you a taste of the language and some of these features. Our goal is to 
choose a representative example so you get a feel for the kinds of programs possible 
with JavaFX. The example (a guitar tuner) illustrates language constructs while keep-
ing the discussion concrete. We•ll veer away from the example at times to illustrate 
additional JavaFX features that are relevant. While this overview is in no way com-
plete (remember, it•s just a taste), we hope to entice you to explore JavaFX further.

The source code for GuitarTuner appears at the end of the chapter (see •Source Code 
for Project GuitarTunerŽ on page 36). To keep the text flowing, we•ll show snippets 
from this application throughout the overview.

What You Will Learn

€ What makes JavaFX unique as a scripting language

€ All about object literals and declarative constructs

€ Introducing the JavaFX scene graph

€ Declaring variables, properties, and objects

€ Initializing objects and object properties

€ Basics in container coordinate space and layout

€ Creating a custom node

€ Manipulating objects with color, effects, and gradients

€ Getting things done with binding, event handlers, and animation

2.1  Introducing JavaFX

What is JavaFX? JavaFX is a scripting language with static typing. You can call a Java 
API as needed from JavaFX and create new object types with classes, but JavaFX also 
provides an easy declarative syntax. (Declarative means you say what you want and 

the system figures out how to do it for you.) JavaFX provides properties for manipu-
lating objects within a 2D coordinate system, specifying fill and pen stroke colors, and 
creating special effects. You can create shapes and lines, manipulate images, play vid-
eos and sounds, and define animations. 

Let•s begin exploring JavaFX by introducing the basics. Our introduction begins with 
project GuitarTuner where you•ll see the main structure of a JavaFX program. Then, 
you•ll explore a few JavaFX language constructs and see how to improve the appear-
ance of your applications. Finally, you•ll see how to make applications do things.

JavaFX in a Nutshell

JavaFX is statically typed, meaning program data types are known at compile time. JavaFX 
also uses type inference. This means you don•t have to declare the type of every variable 
because JavaFX can generally figure it out for you. This gives JavaFX the efficiency of a stati-
cally typed language combined with the ease of a declarative language.

2.2  Project GuitarTuner

Project GuitarTuner helps you tune your guitar. It displays a visual guitar fret board 
with six strings. The letter (note) corresponding to the guitar string appears next to 
the fret board. When you click a string with the mouse, you•ll hear a synthesized gui-
tar note for the selected string as it vibrates visually. Project GuitarTuner uses the Java 
javax.sound.midi  API to generate the sounds. Figure 2.1 shows this application run-
ning when the A string is vibrating. The corresponding JavaFX graphical objects are 
labeled.

Figure 2.1 JavaFX application GuitarTuner

Rectangle (fret board)

GuitarString (6)

Line (2) (frets)

The Scene Graph Metaphor

JavaFX programs with a graphical user interface define a stage and a scene within that 
stage. The stage represents the top level container for all JavaFX objects; that is, the 
content area for an applet or the frame for a widget. The central metaphor in JavaFX 
for specifying graphics and user interaction is a scene graph. A scene defines a hierar-
chical node structure that contains all the scene•s components. Nodes are represented 
by graphical objects, such as geometric shapes (Circle, Rectangle), text, UI controls, 
image viewers, video viewers, and user-created objects (such as GuitarString in our 
example). Nodes can also be containers that in turn hold more nodes, letting you 
group nodes together in hierarchical structures. (For example, Group is a general-pur-
pose container node, HBox provides horizontal layout alignment, and VBox provides 
vertical layout alignment.) The scene graph is this hierarchical node structure. 

Figure 2.2 shows the scene graph for project GuitarTuner. Compare the visual graphi-
cal elements in Figure 2.1 with the scene graph depicted in Figure 2.2. 

In general, to construct a JavaFX application, you build the scene graph, specifying the 
look and behavior of all its nodes. Then, your application just •runs.Ž Some applica-
tions need input to go„user actions that activate animations or affect component 

Scene

Figure 2.2 Nested Scene Graph for GuitarTuner

properties. Other applications just run on their own. (Building the scene graph is anal-
ogous to winding up a toy. When you•re done, the application just runs.)

JavaFX Scene Graph

The power of the scene graph is that, not only do you capture the entire structure of your appli-
cation in a data structure, but you can change the display simply by modifying properties of 
the objects in the scene graph. (For example, if you change a node•s visible  property to false ,
that node, and any nodes it contains, disappears. If you change a node•s location, it moves.)

Within the scene graph for project GuitarTuner, you see the Scene at the top level, 
which contains a Group. Within the Group there is a Rectangle for the fret board (the 
guitar neck), two Line nodes representing frets, and six GuitarStrings. Each Guitar-
String is in turn its own Group consisting of three Rectangles and a Text node. Nodes 
that contain other nodes (such as Scene and Group) include a content  property that 
holds subnodes. The hierarchical nature of the scene graph means that all nodes at the 
same level share the same coordinate space. You therefore build node structures (such 
as GuitarString) that use a relative coordinate system. You•ll see shortly why this is 
useful.

Think Like A Designer

JavaFX encourages you to think like a designer. As a first step, visualize the structure of your 
application or widget and compose your scene out of simple shapes and other building blocks.

The order of nodes within a parent container affects their rendering. That is, the first 
node in the container is •drawnŽ first. The final node is •drawnŽ last and is on top of 
the view. Nodes (depending on their placement within the coordinate system) may 
visually block or •clipŽ previously drawn nodes. In GuitarTuner, the nodes must be in 
a specific order. You draw the fret board first, then the frets, and finally the guitar 
strings, which appear on top. 

Changing the relative order of nodes in a container is easy. The toFront()  function 
brings a node to the front (top) and the toBack()  function sends a node to the back 
(bottom).

Hierarchical Scene Graph

Figure 2.3 also shows a scene graph of project GuitarTuner. Figure 2.2 and Figure 2.3 
depict the same structure, but Figure 2.3 shows the hierarchical relationship among 
the nodes in the scene using a graphical tree view. Nodes at the same level share the 
same coordinate space. For example, the three Rectangles and Text nodes in the Gui-
tarString share the same coordinate system.

2.3  JavaFX Program Structure

JavaFX program structure is simple. For programmers who are used to traditionally 
compiled programs, programming in JavaFX will feel different. With static typing, 
JavaFX gives you feedback at compile time when you use types incorrectly. This 
greatly enhances your ability to write correct code. Furthermore, with the NetBeans 
IDE, you can access JavaDocs for all JavaFX types (classes) and dynamically query 
these class properties and functions, essentially getting feedback at edit time. 

Let•s see how the Stage and Scene form the JavaFX program structure.

Stage and Scene

The Stage is the top-level container and contains the Scene. The Scene, in turn, holds 
nodes that make up the scene graph. Every JavaFX program that has graphical objects 
declares a Stage object.

Here is a top-level implementation of the scene graph for GuitarTuner from Figure 2.2 
(or Figure 2.3). (We•ll look at GuitarString•s node graph shortly.)

Scene

Rectangle Line GuitarString GuitarString. . .

Figure 2.3 Scene Node Graph for project GuitarTuner

Line GuitarString

Group

Rectangle RectangleRectangle Text

(mouse
detection) (normal) (vibrating) (note display)

(frets)(fret board)

GuitarString (CustomNode)

Group

(E) (A) (E)

// Stage and Scene Graph
Stage {

title: "Guitar Tuner" 
Scene {

// content is sequence of SceneGraph nodes
content: [

Group {
content: [

Rectangle { ... }
Line { ... }
Line { ... }
GuitarString { ... }
GuitarString { ... }
GuitarString { ... }
GuitarString { ... }
GuitarString { ... }
GuitarString { ... }

]
} // Group

]
} // Scene

} // Stage

Object Literals

The Stage and Scene objects are instantiated with object literal expressions, or object
literals. Object literals provide a declarative style of programming. Intuitively, declara-
tive means •tell me what you want, not how to do it.Ž As you will see, the real declar-
ative part of JavaFX is binding. We show why this is so powerful later in the chapter.

Object literals require an object (or class) type (such as Stage or Scene) followed by 
curly braces { } . Any properties you need to initialize appear inside the braces. (Stage 
has a title  property and Scene and Group both have content  properties.) Each prop-
erty has a name, followed by a colon : and an initial value for the property. You sep-
arate properties with commas, line breaks, or white space. Here, for example is an 
object literal that initializes a Rectangle (properties x and y designate the upper-left 
corner origin).

Rectangle { x: 10, y: 20, height: 15, width: 150 }

The above Stage, Scene, and Group objects are defined with object literals. Note that 
the Scene object nests inside the Stage object. Likewise, a Group nests inside the 
Scene. Square brackets [ ]  define a sequence of items for the content  property in a 
Scene or Group. Here, the Scene object•s content  property is a sequence of all of the 
top-level nodes of the Scene. In the GuitarTuner application, this is a Group node (see 
Figure 2.2 or Figure 2.3). The Group node likewise includes a content  property with 
all of its subnodes (Rectangles, Lines, and a custom GuitarString). How you nest these 
nodes determines the structure of the scene graph.

Here•s the top-level implementation for GuitarString from its scene graph in 
Figure 2.2 (and Figure 2.3). 

// GuitarString - defined as custom class
Group {

content: [
Rectangle { ... }
Rectangle { ... }
Rectangle { ... }
Text { ... }

]
} // Group

The GuitarString consists of a Group node whose content  property defines a sequence 
containing three rectangles and a Text object. You•ll see how this fits into the Guitar-
Tuner application later on.

2.4  Key JavaFX Features

GuitarTuner is a fairly typical JavaFX example application. It has a graphical repre-
sentation and responds to user input by changing some of its visual properties (as 
well as producing guitar sounds). Let•s look at some of the key JavaFX features it uses 
to give you a broad look at the language.

Signature JavaFX Features

Included in any list of key JavaFX features are binding, node event handlers, and animation. 
We discuss each of these important constructs in their own section (see •Doing ThingsŽ on 
page 31).

Type Inference

JavaFX provides def  for read-only variables and var  for modifiable variables.

def numberFrets = 2; // read-only Integer
var x = 27.5; // variable Number
var y: Number; // default value is 0.0
var s: String; // default value is ""

The compiler infers types from the values you assign to variables. Read-only number-

Frets  has inferred type Integer ; variable x has inferred type Number (Float ). This 
means you don•t have to specify types everywhere (and the compiler tells you when a 
type is required.)

Strings

JavaFX supports dynamic string building. Curly braces { }  within a String expression 
evaluate to the contents of the enclosed variable. You can build Strings by concatenat-
ing these String expressions and String literals. For example, the following snippet 
prints " Greetings, John Doe! ".

def s1 = "John Doe";
println("Greetings, {s1}!"); // Greetings, John Doe!

Shapes

JavaFX has numerous shapes that help you create scene graph nodes. There are 
shapes for creating lines (Line, CubicCurve, QuadCurve, PolyLine, Path) and shapes 
for creating geometric figures (Arc, Circle, Ellipse, Rectangle, Polygon). The Guitar-
Tuner application uses only Rectangle and Line, but you•ll see other shape examples 
throughout this book.

Let•s look at shapes Rectangle and Circle. They are both standard JavaFX shapes that 
extend class Shape (in package javafx.scene.shape) . You define a Circle by specifying 
values for its radius , centerX , and centerY  properties. With Rectangle, you specify val-
ues for properties height , width , x, and y.

Shapes share several properties in common, including properties fill  (type Paint to 
fill the interior of the shape), stroke  (type Paint to provide the outline of the shape), 
and strokeWidth  (an Integer for the width of the outline). 

Here, for example, is a Circle with its center at point (50,50) , radius 30, and color 
Color.RED .

Circle {
radius: 30
centerX: 50 
centerY: 50
fill: Color.RED

}

Here is a Rectangle with its top left corner at point (30, 100) , height 30, width 80, and 
color Color.BLUE .

Rectangle {
x: 30, y: 100
height: 30, width: 80
fill: Color.BLUE

}

All shapes are also Nodes (javafx.scene.Node ). Node is an all-important class that 
provides local geometry for node elements, properties to specify transformations 

(such as translation, rotation, scaling, or shearing), and properties to specify functions 
for mouse and key events. Nodes also have properties that let you assign CSS styles to 
specify rendering.1 We discuss graphical objects in detail in Chapter 4.

Sequences

Sequences let you define a collection of objects that you can access sequentially. You 
must declare the type of object a sequence will hold or provide values so that its type 
can be inferred. For example, the following statements define sequence variables of 
GuitarString and Rectangle objects.

var guitarStrings: GuitarString[];
var rectangleSequence: Rectangle[];

These statements create read-only sequences with def . Here, sequence noteValues  has 
an inferred type of Integer[] ; sequence guitarNotes  has an inferred type of String[] .

def noteValues = [ 40,45,50,55,59,64 ];
def guitarNotes = [ "E","A","D","G","B","E" ];

Sequences have specialized operators and syntax. You will use sequences in JavaFX 
whenever you need to keep track of multiple items of the same object type. The 
GuitarTuner application uses a sequence with a for loop to build multiple Line objects 
(the frets) and GuitarString objects.

// Build Frets
for (i in [0..<numberFrets])

Line { . . . }

// Build Strings
for (i in [0..<numberStrings])

GuitarString { . . . }

The notation [0..<n]  is a sequence literal and defines a range of numbers from 0 to 
n-1 , inclusive.

You can declare and populate sequences easily. The following declarative approach
inserts Rectangles into a sequence called rectangleSequence , stacking six Rectangles 
vertically.

def rectangleSequence = for (i in [0..5])
Rectangle {

1. Cascading Style Sheets (CSS) help style web pages and let designers give a uniform look and 
feel throughout an application, widget, or entire web site. You can use CSS to similarly style 
JavaFX nodes. (See •Cascading Style Sheets (CSS)Ž on page 148 for details on applying styles 
to JavaFX nodes.)

x: 20
y: i * 30
height: 20
width: 40

}

You can also insert number values or objects into an existing sequence using the 
insert  operator. The following imperative approach inserts the six Rectangles into a 
sequence called varRectangleSequence .

var varRectangleSequence: Rectangle[];
for (i in [0..5])

insert Rectangle {
x: 20
y: i * 30
height: 20
width: 40

} into varRectangleSequence;

JavaFX Tip

The declarative approach with rectangleSequence  is always preferred (if possible). By using 
def  rather than var  and declaring sequences rather than inserting objects into them, type 
inference will more likely help you and the compiler can optimize the code more effectively.

You•ll see more uses of sequence types throughout this book.

Calling Java APIs 

You can call any Java API method in JavaFX programs without having to do anything 
special. The GuitarString node •plays a noteŽ by calling function noteOn  found in Java 
class SingleNote. Here is GuitarString function playNote  which invokes SingleNote 
member function noteOn .

function playNote(): Void {
synthNote.noteOn(note); // nothing special to call Java methods
vibrateOn();

}

Class SingleNote uses the Java javax.sound.midi  package to generate a synthesized 
note with a certain value (60 is •middle CŽ). Java class SingleNote is part of project 
GuitarTuner. 

Extending CustomNode

JavaFX offers developers such object-oriented features as user-defined classes, over-
riding virtual functions, and abstract base classes (there is also •mixinŽ inheritance). 

GuitarTuner uses a class hierarchy with subclass GuitarString inheriting from a 
JavaFX class called CustomNode, as shown in Figure 2.4.

This approach lets you build your own graphical objects. In order for a custom object 
to fit seamlessly into a JavaFX scene graph, you base its behavior on a special class 
provided by JavaFX, CustomNode. Class CustomNode is a scene graph node (a type of 
Node, discussed earlier) that lets you specify new classes that extend from it. Just like 
Java, •extendsŽ is the JavaFX language construct that creates an inheritance relation-
ship. Here, GuitarString extends (inherits from) CustomNode. You then supply the 
additional structure and behavior you need for GuitarString objects and override any 
functions required by CustomNode. JavaFX class constructs are discussed in more 
detail in Chapter 3 (see •Classes and ObjectsŽ on page 67).

Here is some of the code from GuitarTuner's GuitarString class. The create  function 
returns a Node defining the Group scene graph for GuitarString. (This scene graph 
matches the node structure in Figure 2.2 on page 15 and Figure 2.3 on page 17. 
Listing 2.2 on page 38 shows the create  function in more detail.)

public class GuitarString extends CustomNode {
// properties, variables, functions
. . . 
protected override function create(): Node {

return Group {
content: [

Rectangle { ... }
Rectangle { ... }
Rectangle { ... }
Text { ... }

]
} // Group

}
} // GuitarString

GuitarString

CustomNode

Figure 2.4 GuitarString Class Hierarchy

Extensible scene
graph node

Encapsulates
graphical structure
and behavior of
GuitarString node

Geometry System 

In JavaFX, nodes are positioned on a two-dimensional coordinate system with the ori-
gin at the upper-left corner. Values for x increase horizontally from left to right and y 
values increase vertically from top to bottom. The coordinate system is always relative 
to the parent container.

Layout/Groups

Layout components specify how you want objects drawn relative to other objects. For 
example, layout component HBox (horizontal box) evenly spaces its subnodes in a 
single row. Layout component VBox (vertical box) evenly spaces its subnodes in a sin-
gle column. Other layout choices are Flow, Tile, and Stack (see •Layout ComponentsŽ 
on page 119). You can nest layout components as needed. 

Grouping nodes into a single entity makes it straightforward to control event han-
dling, animation, group-level properties, and layout for the group as a whole. Each 
group (or layout node) defines a coordinate system that is used by all of its children. 
In GuitarTuner, the top level node in the scene graph is a Group which is centered ver-
tically within the scene. The subnodes are all drawn relative to the origin (0,0)  within 
the top-level Group. Centering the Group, therefore, centers its contents as a whole. 

Benefits of Relative Coordinate Space

Nodes with the same parent share the same relative coordinate space. This keeps any coordinate 
space calculations for subnodes separate from layout issues of the parent container. Then, when 
you move the parent, everything under it moves, keeping relative positions intact. 

JavaFX Script Artifacts

Defining the Stage and Scene are central to most JavaFX applications. However, 
JavaFX scripts can also contain package declarations, import statements, class declara-
tions, functions, variable declarations, statements, and object literal expressions. 
You•ve already seen how object literal expressions can initialize nodes in a scene 
graph. Let•s discuss briefly how you can use these other artifacts.

Since JavaFX is statically typed, you must use either import statements or declare all 
types that are not built-in. You•ll typically define a package and then specify import 
statements. (We discuss working with packages in Chapter 3. See •Script Files and 
PackagesŽ on page 86.) Here is the package declaration and import statements for 
GuitarTuner.

package guitartuner;

import javafx.scene.effect.DropShadow;

import javafx.scene.paint.Color;
import javafx.scene.paint.LinearGradient;

. . . more import statements . . .

import javafx.stage.Stage;
import noteplayer.SingleNote;

If you•re using NetBeans, the IDE can generate import statements for you (type 
Ctrl+Shift+I  in the editor window).

You•ll need script-level variables to store data and read-only variables (def ) for values 
that don•t change. In GuitarTuner, we define several read-only variables that help 
build the guitar strings and a variable ( singleNote ) that communicates with the Java 
midi API. Note that noteValues  and guitarNotes  are def  sequence types.

def noteValues = [ 40,45,50,55,59,64 ];
def guitarNotes = [ "E","A","D","G","B","E" ];
def numberFrets = 2;
def numberStrings = 6;
var singleNote = SingleNote { };

When you declare a Stage, you define the nested nodes in the scene graph. Instead of 
declaring nodes only as object literal expressions, it•s also possible to assign these 
object literals to variables. This lets you refer to them later in your code. (For example, 
the Scene object literal and the Group object literal are assigned to variables in order 
to compute the offset for centering the group vertically in the scene.) 

var scene: Scene;
var group: Group;

scene: scene = Scene { ... }
group = Group { ... }

You may also need to execute JavaFX script statements or define utility functions. 
Here•s how GuitarTuner makes the SingleNote object emit a •guitarŽ sound.

singleNote.setInstrument(27);   // "Clean Guitar"

Once you set up the Stage and scene graph for an application, it•s ready to ready to 
run.2 In GuitarTuner, the application waits for the user to pluck (click) a guitar string. 

2. Java developers may wonder where function main()  is. As it turns out, the JavaFX compiler 
generates a main()  for you, but from a developer•s view, you have just a script file.



26 Chapter 2 A Taste of JavaFX

2.5  Making Things Look Good

Using JavaFX features that enhance the appearance of graphical objects will help your 
application look professionally designed. Here are some simple additions you can 
apply.

Gradients

Gradients lend a depth to surfaces and backgrounds by gradually varying the color of 
the object•s fill  property. In general, use linear gradients with rectangular shapes and 
radial gradients with circles and ovals. In GuitarTuner, the background is a linear gra-
dient that transitions from Color.LIGHTGRAY  (at the top) to the darker Color.GRAY  (at 
the bottom) as shown in Figure 2.5. The guitar fret board also uses a linear gradient.

Here is the LinearGradient for the background scene in GuitarTuner, defined for 
property fill . Note that specifying gradients is declarative; you identify the look you 
want and the system figures out how to achieve it, independent of screen resolution, 
color depth, etc.

fill: LinearGradient {
startX: 0.0
startY: 0.0
endX: 0.0
endY: 1.0
proportional: true
stops: [

Stop {
offset: 0.0
color: Color.LIGHTGRAY 

},
Stop {

offset: 1.0

Figure 2.5 Gradients in the GuitarTuner Application

color: Color.GRAY 
}

]
}

The background gradient changes color along the y axis and the color is constant 
along the x axis (properties startX  and endX are the same). Property stops  is a 
sequence of Stop objects containing an offset  and a color . The offset is a value 
between 0 and 1 inclusive; each succeeding offset must have a higher value than the 
preceding one. 

Property proportional  indicates whether start and end values are proportional 
(defined between [0..1] if true ) or absolute (absolute coordinates if false ).

Radial gradients work well for circular shapes, as shown in Figure 2.6. Here you see 
three Circle shapes, all with radial gradients. The first circle defines a gradient with its 
center in the lower left quadrant ( centerX  is 0.25  and centerY  is 0.75 ). The second cir-
cle•s gradient is centered (centerX  and centerY  are both 0.5 ), and the third circle•s gra-
dient appears in the upper right quadrant ( centerX  is 0.75  and centerY  is 0.25 ).

Here is the radial gradient for the middle circle. 

fill: RadialGradient {
centerX: 0.5 // x center of gradient
centerY: 0.5 // y center of gradient
radius: 0.5 // radius of gradient
stops: [

Stop {
offset: 0
color: Color.WHITE 

},
Stop {

offset: 1

Figure 2.6 Radial Gradients work well with circular shapes

color: Color.DODGERBLUE
}

]
}

Note that the gradient is half the size of the circle (radius  is 0.5 ). Making the gradient 
less than the full size lets the last stop color appear more prominent (the dark color 
predominates).

Color

You specify a shape•s color with property fill . JavaFX has many predefined colors 
ranging alphabetically from Color.ALICEBLUE  to Color.YELLOWGREEN. (In the NetBeans 
IDE, press Ctrl+Space when the cursor is after the dot in Color  to see a complete list, 
as shown in Figure 2.7.)

You can also specify arbitrary colors with Color.rgb  (each RGB value ranges from 0 to 
255),Color.color  (each RGB value ranges from 0 to 1), and Color.web  (a String corre-
sponding to the traditional hexadecimal-based triad). An optional final argument sets 
the opacity, where 1 is fully opaque and 0 is fully translucent. You can also make a 
shape transparent by setting its fill  property to Color.TRANSPARENT.

Here are several examples of color settings. Each example sets the opacity to .5, which 
allows some of the background color to show through.

def c1 = Color.rgb(10, 255, 15, .5);   // bright lime green
def c2 = Color.color(0.5, 0.1, 0.1, .5);    // dark red
def c3 = Color.web("#546270", .5);    // dark blue-gray

Figure 2.7 Explore color choices with the NetBeans IDE

Numeric-based color values (rather than hexadecimal strings or predefined colors) let 
you write functions and animations that numerically manipulate gradients, colors, or 
opacity. For example, the following fill  property gets its Color.rgb  values from a for

loop•s changing value i . The loop produces three different shades of green, depending 
on the value of i .

def rectangleSequence = for (i in [0..2])
Rectangle {

x: 60 * i
y: 50
height: 50
width: 40
fill: Color.rgb(10 + (i*50), 100 + (i*40), i*50)

}

Figure 2.8 shows the resulting set of rectangles with different fill  values.

Rectangles with Arcs

You can soften the corners of Rectangles by specifying properties arcWidth  and 
arcHeight , as shown in Figure 2.9. The first Rectangle has regular, square corners. The 
second Rectangle sets arcHeight  and arcWidth  to 15, and the third one uses value 30

for both. Here•s the object literal for the third Rectangle.

Rectangle {
x: 180
y: 0
height: 70
width: 60
arcHeight: 30
arcWidth: 30
fill: LinearGradient { . . . }

}

Figure 2.8 Manipulating numeric-based Color values

DropShadows

One of the many effects you can specify is DropShadow (effects are declarative). 
Effect DropShadow applies a shadow to its node, giving the node a three-dimensional 
look. In project GuitarTuner, the fret board (guitar neck) uses a default drop shadow, 
as follows.

effect: DropShadow { }

The default object literal provides a drop shadow with these values.

effect: DropShadow {
offsetX: 0.0
offsetY: 0.0
radius: 10.0
color: Color.BLACK
spread: 0.0

}

You can manipulate the location of the shadow by changing offsetX  and offsetY .
Negative values for offsetY  set the shadow above the object and negative values for 
offsetX  set the shadow to the left. Positive values for offsetX  and offsetY  place the 
shadow to the right and below, respectively. You can also change a shadow•s size 
(radius ), color, and spread (how •sharpŽ the shadow appears). A spread value of 1 
means the shadow is sharply delineated. A value of 0 provides a •fuzzyŽ appearance. 
Figure 2.10 shows three rectangles with drop shadows that fall below and to the right 
of the rectangles, using these offsets.

effect: DropShadow {
// shadow appears below and to the right of object
offsetX: 5.0
offsetY: 5.0

}

Figure 2.9 Soften Rectangles with rounded corners

2.6  Doing Things

JavaFX has three main constructs for doing things: binding, node properties that 
define event handlers, and animation. Together, these constructs provide powerful yet 
elegant solutions for modifying scene graphs based on user input or other events. 
Let•s see how GuitarTuner uses these constructs to get its tasks done. 

Binding

Binding in JavaFX is a powerful technique and a concise alternative to specifying tra-
ditional callback event handlers. Basically, binding lets you make a property or vari-
able depend on the value of an expression. When you update any of the •bound toŽ 
objects in the expression, the dependent object automatically changes. Suppose, for 
example, we bind area  to height  and width , as follows.

var height = 3.0;
var width = 4.0;
def area = bind height * width; // area = 12.0

width = 2.5;  // area = 7.5
height = 4;  // area = 10.0

When either height  or width  changes, so does area . Once you bind a property (or vari-
able), you can•t update it directly. For example, you get a compile-time error if you try 
to directly update area .

area = 5; // compile time error

If you make area  a var  and provide a binding expression, you•ll get a runtime error if 
you try to update it directly.

Figure 2.10 Drop shadows provide a three-dimensional effect

In GuitarTuner, the vibrating string changes both its location (property translateY )
and its thickness (property height ) at run time to give the appearance of vibration. 
These properties are bound to other values that control how a guitar string node 
changes.

var vibrateH: Number;
var vibrateY: Number;

Rectangle {
x: 0.0
y: yOffset
width: stringLength
height: bind vibrateH // change height when vibrateH changes
fill: stringColor
visible: false
translateY: bind vibrateY // change translateY when vibrateY changes

}

GuitarTuner also uses bind  to keep the fret board centered vertically by binding prop-
erty layoutY  in the top level group.

group = Group {
layoutY: bind (scene.height - group.layoutBounds.height) /

2 - group.layoutBounds.minY
. . .

}

Node property layoutBounds  provides bounds information for its contents. If a user 
resizes the window, the top level group is automatically centered vertically on the 
screen. Binding helps reduce event processing code because (here, for example) you 
don•t have to write an event handler to detect a change in the window size.

Binding is Good

Binding is good for many things. For example, you can change the appearance of a node based 
on changes to the program•s state. You can make a component visible or hidden. You can also 
use binding to declaratively specify layout constraints. Not only does binding produce less 
code, but the code is less error-prone, easier to maintain, and often easier for the compiler to 
optimize.

Mouse Events

JavaFX nodes have properties for handling mouse and key events. These properties 
are set to callback functions that the system invokes when an event triggers. In Guitar-
Tuner, the •mouse detectionŽ rectangle has the following event handler to detect a 
mouse click event. 

onMouseClicked: function(evt: MouseEvent): Void {
if (evt.button == MouseButton.PRIMARY) {

// play and vibrate selected •stringŽ
}

}

The if  statement checks for a click of the primary mouse button (generally the left 
mouse button is primary) before processing the event. The event handler function 
(shown in the next section) plays the note and vibrates the string.

Animations

JavaFX specializes in animations. (In fact, we dedicate an entire chapter to animation. 
See Chapter 7 beginning on page 205.) You define animations with timelines and then 
invoke Timeline functions play  or playFromStart  (there are also functions pause  and 
stop ). Timelines consist of a sequence of key frame objects that define a frame at a spe-
cific time offset within the timeline. (Key frames are declarative. You say •this is the 
state of the scene at this key timeŽ and let the system figure out how to render the 
affected objects.) Within each key frame, you specify values, an action, or both. Tradi-
tionally, people think of animations as a way to move objects. While this is true, you•ll 
see that JavaFX lets you animate any writable object property. You could, for instance, 
use animation to fade, rotate, resize, or even brighten an image.

Figure 2.11 shows a snapshot of a program with simple animation. It moves a circle 
back and forth across its container.

Here is the timeline that implements this animation using a specialized shorthand 
notation for KeyFrames. The timeline starts out by setting variable x to 0. In gradual, 
linear increments, it changes x so that at four seconds, its value is 350. Now, it per-
forms the action in reverse, gradually changing x so that in four more seconds it is 
back to 0 (autoReverse  is true ). This action is repeated indefinitely (or until the time-
line is stopped or paused). Constants 0s  and 4s  are Duration literals.

Figure 2.11 Timelines let you specify animations

var x: Number;
Timeline {

repeatCount: Timeline.INDEFINITE
autoReverse: true
keyFrames: [

at (0s) { x => 0.0 }
at (4s) { x => 350 tween Interpolator.LINEAR }

]
}.play(); // start Timeline

. . . 
Circle {

. . .
translateX: bind x

}

The JavaFX keyword tween  is a key frame operator that lets you specify how a variable 
changes. Here, we use Interpolator.LINEAR  for a linear change. That is, x doesn•t 
jump from 0 to 350, but gradually takes on values in a linear fashion. Linear interpola-
tion moves the Circle smoothly from 0 to 350, taking four seconds to complete one 
iteration of the timeline.

JavaFX has other interpolators. Interpolator DISCRETE jumps from the value of one key 
frame to the second. Interpolator EASEIN is similar to LINEAR, except the rate of change 
is slower at the onset. Similarly, EASEOUT is slower at the finish and EASEBOTH provides 
easing on both ends of the timeline.

To make this animation apply to the Circle node, you bind the Circle•s translateX

property to the variable manipulated by the timeline ( x). Property translateX  repre-
sents a node•s change in the x direction.

Now let•s examine how GuitarTuner uses animation to vibrate the guitar string and 
play its note. Each GuitarString object uses two rectangles to implement its visible 
behavior. One rectangle is a stationary, thin •stringŽ and represents the string in a 
static state. This motionless rectangle is always visible in the scene. The second rectan-
gle is only visible when the string is •played.Ž This rectangle expands and contracts 
its height quickly using animation (a Timeline). This moving rectangle gives users the 
illusion of a vibrating string. 

To get a uniform vibrating effect, the rectangle must expand and contract evenly on 
the top and bottom. The animation makes the string appear to vibrate by varying the 
height of the rectangle from 1 to 3 while keeping it vertically centered by varying its 
translateY  property between 5 and 4. When the string is clicked, the string•s note 
plays and the rectangle vibrates for the allotted time. When the timeline stops, only 
the stationary rectangle is visible.

Let•s first look at the timeline that plays the note. This timeline appears in the event 
handler for the GuitarString node (see the code for GuitarString in Listing 2.2 on 
page 38).

onMouseClicked: function(evt: MouseEvent): Void {
if (evt.button == MouseButton.PRIMARY) {

Timeline {
keyFrames: [

KeyFrame {
time: 0s
action: playNote // play note and start vibration

}
KeyFrame {

time: 2.8s
action: stopNote // stop playing note and stop vibration

}
]

}.play(); // start Timeline
}

}

Here, the timeline is an object literal defined inside the event handler, invoked with 
function play . This timeline defines a sequence of KeyFrame objects, where function 
playNote  is invoked at time offset 0 seconds and function stopNote  is invoked at time 
offset 2.8 seconds (2.8s ). Here are functions playNote  and stopNote .

// play note and start vibration
function playNote(): Void {

synthNote.noteOn(note);
vibrateOn();

}

// stop playing note and stop vibration
function stopNote(): Void {

synthNote.noteOff(note);
vibrateOff();

}

Function synthNote.noteOn  calls a Java class API to play the guitar string. Function 
vibrateOn  causes the string vibration.

function vibrateOn(): Void {
play.visible = true; // make the vibrating rectangle visible
timeline.play(); // start the vibration timeline

}

Here is the vibration timeline. 

def timeline = Timeline {
repeatCount: Timeline.INDEFINITE
autoReverse: true
keyFrames: [

at (0s) { vibrateH => 1.0 }
at (.01s) { vibrateH => 3.0 tween Interpolator.LINEAR }
at (0s) { vibrateY => 5.0 }
at (.01s) { vibrateY => 4.0 tween Interpolator.LINEAR }

]
};

This timeline uses the shorthand notation discussed earlier for key frames and ani-
mates two variables: vibrateH  and vibrateY . Variable vibrateH  changes the height of 
the rectangle that represents the vibrating string. Variable vibrateY  changes the verti-
cal position of the rectangle to keep it centered as the oscillating height changes. 

2.7  Source Code for Project GuitarTuner

Listing 2.1 and Listing 2.2 show the code for class GuitarString in two parts. 
Listing 2.1 includes the class declarations, functions, class-level variables, and proper-
ties for class GuitarString. Note that several variables are declared public-init . This 
JavaFX keyword means that users of the class can provide initial values with object lit-
erals, but otherwise these properties are read-only. The default accessibility for all 
variables is script-private, making the remaining declarations private.

Use def  for read-only variables and var  for modifiable variables. The GuitarString 
class also provides utility functions that play a note ( playNote ) or stop playing a note 
(stopNote ). Along with the sound, guitar strings vibrate on and off with vibrateOn  and 
vibrateOff . These functions implement the behavior of the GuitarString class.

Listing 2.1 Class GuitarString„Properties, Variables, and Functions

package guitartuner;

import javafx.animation.Interpolator;
import javafx.animation.KeyFrame;
import javafx.animation.Timeline;
import javafx.scene.Cursor;
import javafx.scene.CustomNode;
import javafx.scene.Group;
import javafx.scene.input.MouseButton;
import javafx.scene.input.MouseEvent;
import javafx.scene.Node;
import javafx.scene.paint.Color;
import javafx.scene.shape.Rectangle;
import javafx.scene.text.Font;
import javafx.scene.text.Text;
import noteplayer.SingleNote;

public class GuitarString extends CustomNode {

    // read-only variables
    def stringColor = Color.WHITESMOKE;
    // "Strings" are oriented sideways, so stringLength is the
    // Rectangle width and stringSize is the Rectangle height
    def stringLength = 300;
    def stringSize = 1;
    def stringMouseSize = 15;
    def timeline = Timeline {
        repeatCount: Timeline.INDEFINITE
        autoReverse: true
        keyFrames: [
            at (0s) { vibrateH => 1.0 }
            at (.01s) { vibrateH => 3.0 tween Interpolator.LINEAR }
            at (0s) { vibrateY => 5.0 }
            at (.01s) { vibrateY => 4.0 tween Interpolator.LINEAR }
        ]
    };

    // properties to be initialized
    public-init var synthNote: SingleNote;
    public-init var note: Integer;
    public-init var yOffset: Number;
    public-init var noteText: String;

    // class variables
    var vibrateH: Number;
    var vibrateY: Number;
    var play: Rectangle;

    function vibrateOn(): Void {
        play.visible = true; 
        timeline.play();
    }
    function vibrateOff(): Void {
        play.visible = false;
        timeline.stop();
    }
    function playNote(): Void {
        synthNote.noteOn(note);
        vibrateOn();
    }
    function stopNote(): Void {
        synthNote.noteOff(note);
        vibrateOff();
    }

Listing 2.2 shows the second part of the code for the GuitarString class. 

Every class that extends CustomNode must define a function create  that returns a 
Node object.3 Often the node you return will be a Group, since Group is the most gen-
eral Node type and can include subnodes. But, you can return other Node types, such 
as Rectangle (Shape) or HBox (horizontal box) layout node.

The scene graph for GuitarString is interesting because it actually consists of three 
Rectangle nodes and a Text node. The first Rectangle, used to detect mouse clicks, is 
completely translucent (its opacity  is 0). This Rectangle is wider than the guitar string 
so the user can more easily select it with the mouse. Several properties implement its 
behavior: property cursor  lets a user know the string is selected and property 
onMouseClicked  provides the event handling code (play the note and vibrate the 
string).

The second Rectangle node defines the visible string. The third Rectangle node 
(assigned to variable play ) •vibratesŽ by both moving and changing its height. This 
rectangle is only visible when a note is playing and provides the vibration effect of 
•pluckingŽ a string. The movement and change in height are achieved with animation 
and binding. The Text node simply displays the letter (E, A, D, etc.) associated with 
the guitar string•s note.

Listing 2.2 Scene Graph for GuitarString

    protected override function create(): Node {
        return Group {
            content: [

// Rectangle to detect mouse events for string plucking
     Rectangle {

        x: 0
       y: yOffset

     width: stringLength
    height: stringMouseSize
    // Rectangle has to be "visible" or scene graph will
    // ignore mouse events. Therefore, we make it fully

   // translucent (opacity=0) so it is effectively invisible
     fill: Color.web("#FFFFF", 0)  // translucent
     cursor: Cursor.HAND

    onMouseClicked: function(evt: MouseEvent): Void {
  if (evt.button == MouseButton.PRIMARY){

           Timeline {
            keyFrames: [

    KeyFrame {

3. Well, almost. If you don•t define function create , then you must declare the class abstract .
The Piano example (see •Project PianoŽ on page 167) uses an abstract class.

       time: 0s
action: playNote

                 }
    KeyFrame {

          time: 2.8s
action: stopNote

                 }
 ]  // keyFrames

      }.play();  // start Timeline
          } // if

                    }
   }   // Rectangle

                // Rectangle to render the guitar string
     Rectangle {

        x: 0.0
      y: 5 + yOffset

     width: stringLength
     height: stringSize

      fill: stringColor
                }

// Special "string" that vibrates by changing its height
    // and location

   play = Rectangle {
        x: 0.0

       y: yOffset
     width: stringLength

    height: bind vibrateH
      fill: stringColor
      visible: false

   translateY: bind vibrateY
                }

 Text {      // Display guitar string note name
    x: stringLength + 8

     y: 13 + yOffset
       font: Font {

          size: 20
                    }

      content: noteText
                }
            ]
        }   // Group
    }
} // GuitarString

Listing 2.3 shows the code for Main.fx , the main program for GuitarTuner.

Listing 2.3 Main.fx

package guitartuner;
import guitartuner.GuitarString;
import javafx.scene.effect.DropShadow;
import javafx.scene.Group;
import javafx.scene.paint.Color;
import javafx.scene.paint.LinearGradient;
import javafx.scene.paint.Stop;
import javafx.scene.Scene;
import javafx.scene.shape.Line;
import javafx.scene.shape.Rectangle;
import javafx.stage.Stage;
import noteplayer.SingleNote;

def noteValues = [ 40,45,50,55,59,64 ]; // numeric value required by midi
def guitarNotes = [ "E","A","D","G","B","E" ]; 
// guitar note name
def numberFrets = 2;
def numberStrings = 6;
var singleNote =  SingleNote{};
singleNote.setInstrument(27);   // "Clean Guitar"

var scene: Scene;
var group: Group;
Stage {
    title: "Guitar Tuner"
    visible: true
    scene: scene = Scene {

fill: LinearGradient {
startX: 0.0
startY: 0.0
endX: 0.0
endY: 1.0
proportional: true
stops: [

Stop {
offset: 0.0
color: Color.LIGHTGRAY

},
Stop {

offset: 1.0
color: Color.GRAY

}
]

        }
        width: 340
        height: 200
        content: [
            group = Group {

 // Center the whole group vertically within the scene
 layoutY: bind (scene.height - group.layoutBounds.height) / 



Source Code for Project GuitarTuner 41

2 - group.layoutBounds.minY
     content: [

Rectangle { // guitar neck (fret board)
      effect: DropShadow { }

          x: 0
          y: 0

         width: 300
         height: 121

      fill: LinearGradient {
startX: 0.0
startY: 0.0
endX: 0.0
endY: 1.0
proportional: true
stops: [

Stop {
offset: 0.0
color: Color.SADDLEBROWN 

},
Stop {

offset: 1.0
color: Color.BLACK

}
]

}
} // Rectangle

     for (i in [0..<numberFrets])   // two frets
          Line {

        startX: 100 * (i + 1)
           startY: 0

         endX: 100 * (i + 1)
           endY: 120

         stroke: Color.GRAY
           }

   for (i in [0..<numberStrings])   // six guitar strings
        GuitarString {

        yOffset: i * 20 + 5
         note: noteValues[i]

        noteText: guitarNotes[i]
        synthNote: singleNote

           }
                ]
            }
        ]
    }
}

3 JavaFX Language

Now that you•ve seen JavaFX code in the overview application example, let•s discuss 
the JavaFX scripting language in more detail. This chapter covers the essential aspects 
of the JavaFX language.

What You Will Learn

€ Declaring variables and defining their types

€ Using operators and expressions

€ Using sequences

€ Writing functions with arguments and return values

€ Using function types

€ Defining classes, instance variables, and instance functions

€ Creating objects from classes with object literals

€ Using inheritance and overriding functions

€ Writing init and postinit blocks

€ Using abstract classes and mixin inheritance

€ Using binding and triggers

€ Packaging classes and writing script files

€ Exception handling

3.1  Variables and Types

In JavaFX, you define modifiable variables with the keyword var . Use the keyword 
def  for read-only variables. Attempts to directly modify def  variables result in com-
pile-time errors. 

def maxLength = 100; // read-only

var count = 0; // readable and writable
count++; // count is writable
maxLength = 500; // compiler error

You can also use def  with variables that change through binding (see •Binding 
ExpressionsŽ on page 52). Using def  allows the compiler to generate more efficient 
code and check the correctness of your program.

Programming Tip

If something can be def , it probably should be.

You may have noticed that the above var  and def  statements did not specify any types
for the declarations. This is because JavaFX has a sophisticated inference engine that 
determines types by their use. Type inference makes your code more concise. Of 
course, you can always specify types explicitly if you need to. 

def maxLength: Integer = 100; // read-only
var count: Integer = 0; // readable and writable

JavaFX Types

JavaFX is a statically-typed language with the following built-in types: Boolean , Inte-

ger , Number, String , Duration , and Void . Use Boolean  for boolean variables, Integer  for 
integral variables, Number for floating variables, and String  for strings. Use Void  with 
functions that do not return values. (Note that the Void  keyword begins with a capital 
V in JavaFX.) Duration  types support time literals.

Let•s look at the built-in JavaFX types in more detail.

Boolean

The Boolean  type is handy for variables representing flags and any application-spe-
cific internal state. Use the keywords true  and false  for boolean values. The default 
value for Boolean  is false .

var isElement: Boolean; // Boolean type, default is false
var flag = true; // Boolean type inferred
isElement = 1; // compiler error
flag++; // compiler error

Integer

The Integer  type is signed integral values representing 32 bits. You can use octal (pre-
fix with 0) or hexadecimal notation (prefix with 0x  or 0X) to initialize or assign values. 
The default value for Integer  is 0.

var counter: Integer; // Integer type (32 bits), default is 0
def length = 80; // Integer type inferred
var byte = 0x3f; // hexadecimal value
var word = 037; // octal value

Number

Use Number when you need 32-bit floating-point precision. The default value for Num-

ber  is 0.0 .

var radius: Number; // Number type, default is 0.0
var value = 1.1; // Number type inferred
var big = 1.56e10; // big number
var small = 6.32e-25; // small number

String

The String type represents character strings of any length. You can use double quota-
tion marks ( " ) or single quotation marks ( ' ) to enclose character string literals. Single 
quotation marks are legal within double-quoted strings without being escaped (and 
vice versa). The compiler also merges adjacent string literals for you at compile-time, 
making it convenient to define strings on multiple lines. Use curly braces "{}"  to 
embed expressions in strings. The default value for String  is "" .

var s: String; // String type, default is ""
var a = "datapoint"; // String type inferred
var b = "I can't figure it out, that's too hard";
def c = "Hello " // adjacent strings
"and Goodbye"; // "Hello and Goodbye"
var s1 = 'duck'; // single quotes ok
var s2 = 'soup'; // single quotes ok
var s3 = "{s1} {s2}"; // "duck soup"
var s4 = s1 s2; // compiler error
var s5 = "alpha" "bet"; // "alphabet"

Duration

The Duration  type represents floating values for units of time. Durations are denoted 
with time literals ( ms for milliseconds, s for seconds, m for minutes, and h for hours). 
You typically use Duration  types with animation in JavaFX. The default value for 
Duration  is 0ms.

var timeSlice: Duration; // Duration type, default is 0ms
var period = 100ms; // Duration type inferred, 100 milliseconds
var halfMinute = 30s; // 30 seconds
var halfHour = 30m; // 30 minutes
var halfDay = 12h; // 12 hours
var halfPast = .5h; // half hour

Void

Use the Void  type to define a function that does not return a value.

function startSimulation(): Void {
simulate(100ms);
// no return value

}

(See •FunctionsŽ on page 61 for more details about JavaFX functions.)

But Wait, There•s More . . .

You can also use Java wrapper types in JavaFX, but you probably won•t need them 
very often. These types are primarily used for interoperability with Java classes that 
use primitives in their interfaces. Use Character for 16-bit Unicodes, Byte  for 8 bits, 
Short  for 16 bits, Integer  for 32 bits, and Long  for 64 bits (all signed integral values). 
For floating values, use Float  (32 bits) or Double  (64 bits). In JavaFX, the Number type is 
Float  (32 bits). Here are some examples.

var ch: Character = 120; // Unicode character (16 bits) - 'x'
var ts: Byte = 127; // 8 bits
var data: Byte = 500; // compiler error - too large for 8 bits
var ts: Short = 10; // Short (16 bits)
var sdata: Short = 50000; // compiler error - too large for 16 bits
var tl: Long = 100000; // Long (64 bits)
var tg: Double = 3.45; // Double (64 bits)
var tf: Float = 1.23; // Float type (32 bits) - same as Number

Printing Variables

Printing the values of JavaFX variables on the console screen is easy. The function 
print( var )  prints the value of var  and println( var )  prints the value of var  followed 
by a newline. Here are some examples.

var b = true;       // Boolean type
println(b);         // true
var i = 12;       // Integer type
println(i);          // 12
var f = 3.45;       // Number type
println(f); // 3.45
var s = "javafx"; // String type

println(s); // javafx
var d = 100ms; // Duration type
println(d); // 100.0ms
var a: Character = 97; // Character type
println(a); // 'a'
print("last");          // last (no newline)
println("");         // empty string with newline

You can also use curly braces "{}"  to embed variable values in string expressions.

var i = 12; var s = "javafx";
println("i = {i}, s = {s}");    // i = 12, s = javafx

For more sophisticated formatting, •printf-likeŽ formats are available within "{}" .

var i = 12; var s = "javafx";
println("i = {%3d i}, s = {%.4s s}"); // i =  12, s = java
var v = 56.789;
println("v = {%8.2f v}");    // v =    56.79

The first println()  prints integer i  right-justified in a field of 3 characters followed by 
the first four characters of the string s. The second println()  prints floating value v in 
a field of 8 characters with 2 digits to the right of the decimal point. Refer to the Java-
doc for java.lang.System.printf()  to learn more about these formatting options.

Pseudo Variables

JavaFX supports several pre-defined variables that you can use in your programs. 
These variables are script-level read-only def  variables.

Table 3.1 shows the JavaFX pseudo-variables. 

3.2  Operators

JavaFX has arithmetic, assignment, unary, relational, and logical operators. The 
instanceof  operator determines variable types. Let•s look at each of these operators 
separately.

TABLE 3.1 Built-In JavaFX pseudo variables

Name Description

__PROFILE__ Environment (•mobileŽ, •desktopŽ, •browserŽ)

__FILE__ Script file name (full path name)

__DIR__ Script file directory (full path name)

Arithmetic Operators

The arithmetic operators are add (+), subtract (- ), multiply ( * ), divide ( / ) and divide 
with integer remainder ( mod). Here are several examples with Integer  variables.

var val = 25; var nt = 4;
var a = val + nt; // a is 29
var b = val - nt; // b is 21
var c = val * nt; // c is 100
var d = val / nt; // d is 6
var e = val mod nt; // e is 1

Mixed type arithmetic will do what is reasonable, as these examples show. 

var fv = 6.5; // fv is Number type
var ic = 6; // ic is Integer type
var td = 100ms; // td is Duration type
var tg = 10ms; // tg is Duration type

var gv = fv * ic; // gv is Number type
var tf = td / fv; // tf is Duration type
var hm = td / tg; // hm is Number type
var mm = td * tg; // compiler error

Assignment Operators

Besides conventional assignments with =, you can also form compound assignments 
with arithmetic operators. Here are some examples with Integer  variables.

var val = 25; var nt = 4;
val += nt; // val = val + nt; (val is 29)
val -= nt; // val = val - nt; (val is 25)
val *= nt; // val = val * nt; (val is 100)
val /= nt; // val = val / nt; (val is 25)

The mod operator is not allowed in a compound assignment. Instead, you must do

val = val mod nt; // val is 1

You can also assign a sequence of values with [] .

var summerMonths = ["June", "July", "August"];
var powersOfTen = [10, 100, 1000, 10000];

For more on assignments with sequences, see •Sequence LiteralsŽ on page 54.

Unary Operators

Although most operators in JavaFX are binary (two operands), there are several unary 
(one operand) operators. These are ++ (preincrement, postincrement), --  (predecre-

ment, postdecrement), -  (negation), and not  (logical complement). Here are some 
examples.

var data = 10;
var negate = -data; // negate is -10
var m; var n;
var s = 5; var t = 5;
m = s++; // m is 5, s is 6
n = ++s; // n is 7, s is 7
m = t--; // m is 5, t is 4
n = --t; // n is 3, t is 3
var enabled = false;
var start = not enabled; // start is true

Relational Operators

The relational operators compare values (>, >=, <, <=) and test for equality (==, != ).
Here are several examples.

var f = 10.5; var g = 20.5; var h;
h = (f == g); // h is false
h = (f != g); // h is true
h = (f > g); // h is false
h = (f < g); // h is true
h = (f >= g); // h is false
h = (f <= g); // h is true
h = (f > 10); // h is true

Note that the inferred type is Boolean for variable h in these examples. The last exam-
ple compares different types (Number and Integer ). In this case, the Integer  is first con-
verted to Number before the comparison.

With Strings, the equality operators ( ==, != ) in JavaFX perform value comparisons (this 
is different than Java).

def input: String = getInput(); // get input String
var exit = (input == "quit"); // true if input string is "quit"

Logical Operators

The and operator and the or  operator let you combine boolean expressions. 

var i = 2; var j = 3; var k = 4;
var p = (j > i and j < k); // p is true
var q = (j > k or i < k); // q is true

Both operators have •short-circuitŽ behaviors; that is, if the first expression of opera-
tor and  is false, the second expression is not evaluated. Likewise, the second expres-
sion of operator or  is not evaluated if the first expression is true.

Instanceof Operator

The instanceof  operator tells you if an object is a specific type, returning true  for a 
correct type match. Here are several examples.

def u = "string of chars"; // String
var v = u instanceof String; // v is true
def w = 6.5; // Number (Float)
v = w instanceof java.lang.Float; // v is true

You typically use instanceof  with the as  operator to determine if an object is an 
instance of a specific class at runtime. (See •As OperatorŽ on page 80.)

3.3  Expressions

JavaFX is an expression language with value expressions. Many forms of blocks, loops, 
and conditionals in JavaFX are actually expressions in themselves with a value and a 
type. In certain cases, for instance, you can use the return value of an if  statement in 
an assignment statement or the return value from a for  loop to create a new sequence. 
You can also bind expressions. 

Let•s look at JavaFX expressions in more detail.

Block Expressions

A block in JavaFX is a list of expressions enclosed by curly braces. The value of the 
block expression is the value of the last expression. Blocks can be empty (their type is 
Void ) and var  and def  statements are expressions, too (their value is the value of the 
new variable).

Typically, blocks are used for function bodies and with if  statements, for  loops, and 
while  loops. Blocks can have local var  variables and def  constants whose scope is 
defined only within the block where they are declared. However, blocks can also 
return values in JavaFX.

var a = 10; var b = 20; var c = 30;
var sum = { // start of block

def d = 40; def e = 50;
a + b + c + d + e // value of last expression

} // end of block
println(sum);     // sum is 150

A block•s type and value is the type and value of the last expression in the block, 
which does not require a terminating semicolon.

If Expressions

An if  expression alters program flow based on the value of a boolean expression as a 
conditional.

var num;
var i: Integer = getNumber();
if (i >= 0 and i <= 9)

num = "single digit";

The else  keyword can also be used with if  expressions.

var num;
var i: Integer = getNumber();
if (i >= 0 and i <= 9) {

num = "single digit";
} else if (i >= 10 and i <= 99) {

num = "two digits";
} else

num = "three digits or more";

Sinceif  expressions are value expressions in JavaFX, we can collapse all this code into 
the following terse form (no braces needed).

var i: Integer = getNumber();
var num = if (i >= 0 and i <= 9) "single digit" 

else if (i >= 10 and i <= 99) "two digits" 
else "three digits or more";

Note that if  controls the execution of one expression. If that expression includes one 
statement, a block with braces is optional. If that expression includes more than one 
statement, you must use a block.

For Expressions

A for loop is an easy way to do something a certain number of times.

for (i in [1..5]) {
doSomething(); // call function five times

}

The notation [1..5]  generates a sequence literal (see •Sequence LiteralsŽ on page 54).

A for loop is also convenient for going through a sequence of items.

var strings = ["straight", "forward", "thinking"];
for (s in strings) {

writeText(s); // call function three times with s
}

This for loop fetches three strings from a sequence, one at a time, assigns each one to 
variable s, and passes it as an argument to function writeText .

Programming Tip

Note that you do not declare loop variables with var  or def  before the keyword in . You cannot 
modify a loop variable and its scope is the loop where it is defined.

A for loop is actually an expression that returns a sequence. We show you how to use 
this powerful technique in •Creating Sequences with forŽ on page 56.

While Expressions

A while  expression loops until a given condition is false. While expressions do not 
return a value. While loops are typically used when you can easily express the loop 
continuation condition as a boolean. Here is an example.

var u = 1234; // count number of digits in number
var numDigits = 0;
while (u != 0) {

numDigits++;
u /= 10;

}
println(numDigits); // numDigits is 4

Break and Continue

The keywords break  and continue  may only appear inside for  and while  expressions. 
Use continue  to skip to the next iteration and break  to exit a loop immediately.

var str: String = getString();
for (weekDay in ["Mon", "Tues", "Wed", "Thu", "Fri"]) {

if (weekDay == "Wed") continue;    // skip Wed
. . .
if (str == weekDay) break;         // exit loop

}

Binding Expressions

One of the most powerful features of JavaFX is binding. The idea is to bind a variable 
in your program to something that changes when your program runs. This creates a 
dependency between your variable and what is changing. When dependencies change, 
your variable automatically updates. JavaFX lets you bind def  or var  variables to 
expressions (blocks, conditionals, functions, for loops, and object literals).

The format for binding expressions is

def v = bind expression;

When expression  changes, v is updated.

Let•s look at several examples. Suppose you are calculating a product between two 
numbers. When either number changes, you want the product value to stay current. 
Binding the product expression does the job.

var a = 10; var b = 2;
def product = bind a * b; // bind to expression
println(product);         // 20
a = 100;
println(product);        // 200
b = 5;
println(product);        // 500
product = 10;      // compiler error

The compiler lets you declare bound variables with var , but you cannot modify them 
(see •Overriding bindŽ on page 75). If you try, you get a runtime error rather than a 
compiler error.

var prod = bind a * b; // bind to expression
prod = 10;        // AssignToBoundException at runtime

Here•s an example that calls a getNumber  function to update variable top . Binding the 
if  expression insures that top  gets a value no greater than 100.

def top = bind if (num <= 100) num else 100; // bind to conditional
var num = getNumber(); // suppose num is 50
println(top); // top is 50
num = getNumber(); // suppose num is 500
println(top); // top is 100

Here•s another example that binds the addition of variables in a block expression to a 
sum variable.

a = 10; b = 20; var c = 30;
def sum = bind { // bind to block expression

def d = 40; def e = 50;
a + b + c + d + e // binds to this statement

}
println(sum);     // sum is 150
b = 50; c = 50;
println(sum);    // sum is 200

Recall that the value of a block expression is the value of its last expression. With bind ,
all other expressions in the block must be defined with def  or var .

Bidirectional Binding

Bidirectional binding is often used with user interactions to make sure that stored val-
ues and displayed values are in sync. The format for bidirectional binding is

var v = bind w with inverse ;

This format allows updates to occur in both directions. Variable v updates if w

changes, and vice versa. Note that var  must be used with bidirectional binding, since 
v is modifiable.

Here•s an example that keeps two variables (field  and name) in sync.

var field = "one";
var name = bind field with inverse; // name is "one"
field = "two";      // name is "two"
name = "three";     // field is "three"

3.4  Sequences

A sequence is an ordered list of items in JavaFX, similar to arrays in other languages. 
Sequences are very powerful in JavaFX, because you can use them to store any data 
type, including objects. In this section, we show you how to create sequences and 
access sequence items, as well as insert items into sequences and delete items from 
sequences. We also discuss comparing sequences, sequence slices, predicates, and 
binding sequences.

Sequence Literals

Sequence literals are created with a comma-separated list of items within [] .

def primeNumbers = [2, 3, 5, 7, 11, 13]; // Integer values
var brothers = ["groucho", "chico", "harpo"]; // String values

When you declare a sequence, use var  if you intend to modify the sequence; otherwise 
use def . The JavaFX compiler can usually determine sequence types by inference, but 
you can include an explicit type (using [] ) if it•s necessary (the compiler will let you 
know when you need to do this). 

def primeNumbers: Integer[] = [2, 3, 5, 7, 11, 13];
var brothers: String[] = ["groucho", "chico", "harpo"];

There is a convenient notation in JavaFX for generating sequences of values. 

def nineties = [1990..1999]; // [1990, 1991, 1992, ..., 1999]

The notation [ v.. w]  generates a sequence literal with an increasing range of values, 
where v and w are Integer  or Number types. Each new value in the range is one more 
than the previous one. Here are some more examples.

var g = [1..5]; // same as [1, 2, 3, 4, 5]
var h = [1.1..5.1]; // same as [1.1, 2.1, 3.1, 4.1, 5.1]
var j = [1..<5]; // same as [1, 2, 3, 4];

var k = [1..5.0]; // same as [1.0, 2.0, 3.0, 4.0, 5.0]
var m = [4.5..7]; // same as [4.5, 5.5, 6.5]
var n = [5..1]; // no incrementing values, same as []

The default increment for each new range number is 1, but you can use the notation 
[ v..w step n]  to step by different values (n is a positive or negative Integer  or Number

type). This makes it easier to generate more arbitrary sequences.

var p = [0..9 step 2]; // same as [0, 2, 4, 6, 8]
var q = [5..1 step -1]; // same as [5, 4, 3, 2, 1]
var r = [1..3 step .5]; // same as [1.0, 1.5, 2.0, 2.5, 3.0]

The sizeof  operator gives you the length of a sequence.

def primeNumbers = [2, 3, 5, 7, 11, 13];
var length = sizeof(primeNumbers); // length is 6
var length = sizeof primeNumbers; // same thing

The reverse  operator reverses items in a sequence.

var brothers = ["groucho", "chico", "harpo"];
var revBros = reverse brothers; // ["harpo", "chico", "groucho"]

And, if you nest sequences, you get a •flattenedŽ sequence.

def boys = ["billy", "joey"];
def girls = ["mary", "susie"];
var kids = [boys, girls]; // ["billy", "joey", "mary", "susie"];

Printing Sequences

You can use print()  or println()  directly to print all values of a sequence.

def primeNumbers = [2, 3, 5, 7, 11, 13];
println(primeNumbers);         // [ 2, 3, 5, 7, 11, 13 ]
println("{primeNumbers}");   // 23571113

Programming Tip

A sequence name (primeNumbers ) with println()  makes the sequence values easy to read for 
debugging. In contrast, the string expression "{primeNumbers}"  with println()  displays a 
string of characters.

If you need to print sequence elements individually, use a for loop.

def primeNumbers = [2, 3, 5, 7, 11, 13];
for (n in primeNumbers) {

    print("{n} ");
}
println(""); // 2 3 5 7 11 13

The indexof  operator gives you a sequence item•s integer index number (zero-based). 
This operator can only be applied to a loop induction variable. Here is an example 
using the indexof  operator in a for loop. 

var colors = ["red", "blue", "green"];
for (c in colors) {

println("{c} is color #{indexof c}");
}

Each iteration of this for loop displays a color name and the index number (zero-
based) in the sequence.

red is color #0
blue is color #1
green is color #2

Creating Sequences with for

A for  expression returns a sequence, so you can use for loops to create new 
sequences. Here are some examples.

var oddNumbers = for (n in [1..9 step 2]) n; // [1, 3, 5, 7, 9]
var cubes = for (n in [1..5]) n * n * n; // [1, 8, 27, 64, 125]

The following example generates a sequence of integers in a for loop (a pair at a time), 
where the second integer is twice the value of the first integer. (Bracket notation []  is 
required in the body of the for loop here.)

def seq = for (n in [2..6]) [n, 2*n];  // [2, 4, 3, 6, 4, 8, 5, 10, 6, 12]

A for  expression can have more than one in  clause. Here•s an example that calculates 
a sequence of cross products.

var row = [10, 100];
var col = [5, 15];
var crossProduct = for (r in row, c in col) r * c;
println(crossProduct); // [ 50, 150, 500, 1500 ]

Note that the for  expression acts like two for loops with loop variable r  in the outer 
loop and loop variable c in the inner loop.

A for  expression can also have a where  clause, which must be a boolean expression. 
This constrains selections, as the following example shows. 

var numbers = for (r in row, c in col where r * c < 500) r * c;
println(numbers);           // [ 50, 150 ]

Here•s a variant of the same example.

for (r in row, c in col where r * c < 500) {
print("r = {r} c = {c}, ");    // r = 10 c = 5, r = 10 c = 15, 

}

Accessing Sequence Items

Use the expression seq [ index ]  to access items in sequence seq , where index  is zero-
based. If index  is outside the bounds of the sequence (positive or negative), the expres-
sion evaluates to the default value for that item•s type.1

Here are several examples.

var brothers = ["groucho", "chico", "harpo"];
var firstBro = brothers[0]; // firstBro is "groucho"
var lastBro = brothers[2]; // lastBro is "harpo"
var noBro = brothers[3]; // noBro is ""
var nonBro = brothers[-1]; // nonBro is ""

You can also use for loops with sizeof  to access or modify sequence elements.

var list = [1, 2, 3, 4];
for (i in [0..<sizeof(list)]) {

    print(list[i]);
}
println(""); // 1234

for (i in [0..<sizeof(list)]) {
    list[i] = -1;       // set each element to -1

}

Here are some alternative ways to set sequence elements to specific values with for .

for (n in list) list[indexof n] = -1; // set each element to -1
list = for (n in list) {-1}; // set each element to -1

Inserting Items into Sequences

The insert  keyword lets you insert new items into sequences. You may insert items at 
the end of a sequence (insert into ), before an indexed element (insert before ), or 
after an indexed element (insert after ). Here are some examples.

var colors = ["red", "blue"];
insert "green" into colors; // ["red", "blue", "green"]
insert "pink" after colors[1]; // ["red", "blue", "pink", "green"]
insert "cyan" before colors[0]; // ["cyan", "red", "blue", "pink", "green"]

Inserting before or after a negative indexed element inserts at the beginning of the 
sequence. Inserting before or after an indexed element beyond the end of a sequence 
inserts at the end.2

1. This is an example of •The show must go onŽ philosophy of JavaFX.

2. Again, •The show must go on.Ž

Programming Tip

Assigning sequences copies values. Look at the following example.

var s = [1..5];
var t = s; // makes a copy
insert 6 into s;
println(t); // [ 1, 2, 3, 4, 5 ]
println(s); // [ 1, 2, 3, 4, 5, 6 ]

This shows that values are copied when you assign a sequence to another sequence. The same is 
true when you pass a sequence as an argument to a function.

Here•s a for loop that inserts Circles into a sequence.

var theCircles: Circle[];
for (i in [1..3])

insert Circle {
radius: 5
fill: Color.RED

} into theCircles;

Here•s a more efficient way to build a sequence of Circles using the expression-nature 
of for loops. 

def theCircles = for (i in [1..3])
Circle {

radius: 5
fill: Color.RED

}

Deleting Items from Sequences

The delete  keyword lets you delete items from sequences. Table 3.2 shows you the 
different delete formats.

Here are some examples of delete .

var bros = ["groucho", "chico", "harpo", "zeppo", "gummo"];

TABLE 3.2 Sequence Delete Formats 

delete sequence [ index ] Delete item at index  of sequence . If no element at 
that index, ignore the delete

delete sequence [ low..high ] Delete the items from low  to high  inclusive (slice)

delete value  from sequence Delete all occurrences of value  from sequence . If 
no value  found, ignore the delete

delete sequence Delete all elements from sequence  but not the 
sequence itself (sequence is empty)

delete bros[4]; // delete "gummo"
delete bros[2..3]; // delete "harpo" and "zeppo"
delete "chico" from bros; // delete "chico"
delete bros; // delete "groucho" (empty sequence)

Comparing Sequences

Use the == and !=  relational operators to compare sequences. Two sequences are equal 
if they have the same values for all items and they have the same length. Both compar-
ison expressions return boolean values (true or false). Here are some examples.

var sample = [1, 3, 5, 7, 9];
var a = (sample == [1..10 step 2]); // a is true
var b = (sample == [1..5 step 2]); // b is false
var c = (sample != [1..10 step 3]); // c is true

The other relational operators (<, <=, >, >=) cannot be used with sequences.

Sequence Slices

The expression sequence [ low .. high ]  is called a sequence slice. The values for low  and 
high  (Integer  or Number types) are the beginning and ending of a sequence range 
(inclusive on both ends). Here are several examples.

var bros = ["groucho", "chico", "harpo", "zeppo", "gummo"];
var funnyBros = bros[0..2]; // ["groucho", "chico", "harpo"]
var otherBros = bros[3..]; // ["zeppo", "gummo"]
var movieBros = bros[0..<4]; // ["groucho", "chico", "harpo", "zeppo"]
bros[3..4] = ["Z", "G"]; // ["groucho", "chico", "harpo", "Z", "G"]

Predicates

A predicate is a boolean expression that helps generate new sequences as subsets of 
existing sequences. A predicate is used with sequences in JavaFX as follows.

seq[ name | BooleanExpression ]

Here, seq  is the source sequence, name is the selection variable, |  is the predicate sepa-
rator, and BooleanExpression  is the constraint. Elements are selected from the source 
sequence if and only if the constraint is true.

The following example generates a sequence of numbers whose square is less than 50.

def seq = [1..100];
def newSeq = seq[n | (n*n) < 50]; // [1, 2, 3, 4, 5, 6, 7]

Here, n is the selection variable from the source seq . An item is only selected for the 
new sequence (newSeq) if its square (n*n ) is less than 50.

Note that you can use a for loop with a where  clause to generate the same sequence 
without a predicate.

def newSeq = for (n in [1..100] where (n*n) < 50) {n};

Binding Sequences

The format for binding sequences is

def newSeq = bind seq;

If seq  changes, newseq  is updated. Here•s an example.

var seq = [1, 2, 3];
def myseq = bind seq;    // bind to sequence
insert 4 into seq;
println(myseq);          // [ 1, 2, 3, 4 ]
delete 2 from seq;
println(myseq);           // [ 1, 3, 4 ]

In a graphical context, binding property content  keeps a scene graph current. Here, a 
Group object literal binds its content  property to sequence nodeSequence . Changes to 
nodeSequence  updates the Group•s content .

Group {
content: bind nodeSequence
. . .

}

You can also bind for  expressions, since they return sequences. The format is

def newSeq = bind for ( elem  in seq ) expression ;

If sequence seq  changes, a new sequence newSeq is created. Existing sequence elements 
are not recalculated unless you insert or delete elements.

Here•s an example that updates a scaled sequence of numbers.

var scale = 10;
def t = bind for (item in [1..5]) item * scale; // bind to for expression
println(t);           // [ 10, 20, 30, 40, 50 ]
scale = 100;
println(t);          // [ 100, 200, 300, 400, 500 ]

The for loop multiplies five numbers in a sequence by a scale  variable and is bound to 
variable t . If variable scale  changes, sequence t  is updated. 

3.5  Functions

Functions help you centralize code from different places in your scripts. Functions can 
take arguments and can return values. You can also bind functions, pass program 
arguments to scripts, and use functions as special types.

Defining Functions

The keyword function  defines a function and its block of code. If the function doesn•t 
return anything, use the keyword Void  for its return type. Here is a function that 
prints a message when it has been called.

function myPrint(): Void {
println("myPrint called");

}

Passing Arguments to Functions

To pass arguments to functions, include the names and types in the function•s signa-
ture. Here function printStrings  prints its two string arguments. The function doesn•t 
return anything ( Void  return type).

function printStrings(a: String, b: String): Void {
println("{a}{b}");

}

printStrings("alpha", "bet"); // prints "alphabet"

Of course, JavaFX•s inference engine can deduce argument types and return types for 
you, so here is a more compact and general version of printStrings .

function printStrings(a, b) {
println("{a}{b}");

}

printStrings("alpha", "bet"); // prints "alphabet"

One important point about JavaFX function parameters: they are passed by value
inside functions. This makes it impossible to write certain functions in JavaFX, like the 
following swap function which attempts to swap the values of its arguments.

function swap(a, b) {
    var tmp = a;
    a = b;              // won't compile
   b = tmp;            // won't compile

}

Programming Tip

If you pass a sequence to a function as an argument, you cannot change its elements in the 
function (this is different from Java arrays).Object references are also passed by value (see 
•Classes and ObjectsŽ on page 67).

Returning Values from Functions

If a function returns a value, you can include its return type when you define the func-
tion. Here•s a concat  function that concatenates its parameters into a new string and 
returns it.

function concat(a: String, b: String): String {
return "{a}{b}";

}
var s = concat("set", "up"); // s is "setup"

Of course we can rely on inference to write a more compact version of concat .

function concat(a, b) {
return "{a}{b}";

}
var s = concat("alpha", "bet"); // s is "alphabet"

These have all been simple examples, so let•s show you a more interesting one. The 
following gcd  function calculates the greatest common denominator from two inte-
gers passed to the function.

function gcd(x: Integer, y: Integer) {
    var a = x; var b = y;
    while (b != 0) {

var t = a;
a = b;
b = t mod a;

    }
    return a;

}

Here•s a for loop to try out gcd .

for (n in [1..10]) {
    println("gcd of 12 and {n} is {gcd(12, n)}");

}

The output looks like this.

gcd of 12 and 1 is 1
gcd of 12 and 2 is 2
gcd of 12 and 3 is 3
gcd of 12 and 4 is 4
gcd of 12 and 5 is 1

gcd of 12 and 6 is 6
gcd of 12 and 7 is 1
gcd of 12 and 8 is 4
gcd of 12 and 9 is 3
gcd of 12 and 10 is 2

Binding Function Calls

In addition to binding expressions, you can also bind function calls. The format for 
binding function calls is

def v = bind function(arg1, arg2, ...);

The binding concept is similar to binding expressions. When any of the arguments 
(arg1, arg2, ... ) change in the function call, v is updated with the value returned by 
the function.

Let•s show you an example. Suppose you have a hypot  function that calculates the 
hypotenuse for right triangles. Here•s an example that binds the hypot  function to 
variable hypotenuse  to keep its value current when the base or height change.

import java.lang.Math;

function hypot(a, b) {
return Math.sqrt(a * a + b * b);

}
var base = 3; var height = 4;
def hypotenuse = bind hypot(base, height); // bind to function call
println(hypotenuese);             // 5.0
base = 30; height = 40;
println(hypotenuse);             // 50.0

Bound Functions

It may not be sufficient to bind to a function and update only when the function argu-
ments change. You may want to update when variables in the function body change, 
too. To do this, you need a bound function. The format for a bound function is

bound function functionName ( arg1 , arg2 , ...) { . . . }

All expressions in the function body must be defined with def  or var . If you call 
bound functions in a non-bind context, they are ordinary functions.

To see how useful bound functions are, here is an example that uses a bound function 
getTotal  to update the contents of a Text object when selections are made with GUI 
components.

bound function getTotal(): String {
def item1 = if (itemOne.selected) .75 else 0;
def item2 = if (itemTwo.selected) .50 else 0;

def item3 = if (itemThree.selected) .25 else 0;
def total = item1 + item2 + item3;
return "$ {total}";

}

def finalOrder = Text {
content: bind getTotal() // must use bind here
font: Font {

size: 18
}

}

The Text content  property is automatically updated to a new total when any of the 
selections (itemOne.selected , itemTwo.selected , itemThree.selected ) inside the 
getTotal  function body change.

Think of bound functions as •black boxes.Ž The dependencies of an ordinary function 
call in bind context are the dependencies of its arguments. Bound functions calculate 
the dependencies dynamically, based on what actually goes into the calculation.

Bound functions are not automatically treated as if their result is bound. You need to 
use bind (as shown in the above example).

Program Arguments

In certain situations (applets, command line scripts) you may want to pass arguments 
to your JavaFX programs at runtime. These program arguments are available to your 
JavaFX programs as strings. To access program arguments, you can specify a run  func-
tion.3

To demonstrate, here•s a program that echoes its program arguments to the console 
screen in reverse order.

// Echo.fx - Echo program arguments in reverse
function run(args: String[]) {

for (arg in reverse args)
print("{arg} ");

    println("");
}

$ javafxc Echo.fx
$ javafx Echo one two three
three two one

The program arguments are passed to run  as a sequence of Strings. The for loop calls 
reverse  to create a new sequence with the strings in reverse order. Inside the for loop, 

3. Like main  in Java, run  is a magic name.

the print  function prints each argument from this sequence to the screen followed by 
a space. Outside the for loop, the println  supplies the newline.

Programming Tip

You can also call FX.getArguments  to access program arguments. This approach does not 
require a run  function.

var args = FX.getArguments();
for (arg in reverse args)

print("{arg} ");
println("");

Function Types

A more advanced topic with JavaFX functions is function types. In addition to return-
ing built-in types and class types, JavaFX functions can return a function, or more spe-
cifically, a function literal type. Why would you want to do this? Using a technique 
called closures, function literals make libraries more reusable. Let•s investigate this 
technique and show you how to use it.

Suppose you are interested in scaling data in your application; that is, multiplying 
variables in your program by scaling factors at run time. Let•s look at a technique that 
sets up this concept in three steps. The first step is to define a function, as follows.

function scale(n) {
    function(x) { x * n } // function literal

}

The interesting thing about this scale  function is that it doesn•t return a value. 
Instead, scale  accepts a scaling factor (n) and returns a function literal that multiplies 
its parameter (x) by n. Note that this function literal (an anonymous function) is not
called when scale  is invoked.

The second step is to call the scale  function with a scaling factor ( 100) and assign the 
returned function literal to a variable.

var scaleHundred = scale(100); // returns function literal

Now you can use scaleHundred  to call the anonymous function in the function literal. 
Think of it as a pointer or reference to a function that accepts a single argument. 

The third step is to use the scaleHundred  variable to call the anonymous function in 
the function literal with a variable in your program.

var m = 5;
println(scaleHundred(m)); // prints 500

Note that in order to make this work, the scale  function must •rememberŽ the scaling 
factor (100) that was passed to it in the second step. This construct is called a closure.4

When the function in the function literal is invoked with an argument in step three, 
the value of this argument is multiplied by the scaling factor. Of course we can always 
change the scaling factor at run time, which is the point of all this.

scaleHundred = scale(200); // change scaling factor to 200
println(scaleHundred(m)); // prints 1000

To see what•s going on under the hood, here is the scale  function again, fully defined.

function scale(n: Number): function(x: Number): Number {
return function(x: Number): Number { x * n }

}

The scale  function accepts a Number argument and returns a function literal. This func-
tion literal takes a Number argument and returns a Number type, which is the new scaled 
value. Fortunately, type inference in JavaFX makes explicit type declarations like this 
unnecessary in most cases.

All this seems like a lot of work just to multiply two values. The real power of function 
types and closures is with sequences. Here is an example that multiplies all the values 
of a sequence by a scaling factor of 10.

function mult(n: Integer) {
function(seq: Integer[]) {

var thisSeq = seq;
for (i in thisSeq) { thisSeq[indexof i] *= n; }

}
}

var seq = [1, 2, 3, 4, 5]; // original sequence
var multTen = mult(10); // scaling factor is 10
var newSeq = multTen(seq); // generate new sequence
println(newSeq); // [ 10, 20, 30, 40, 50 ]

The mult  function accepts a scaling factor as an argument and returns a function literal 
(an explicit return  is not necessary here). This function literal takes a sequence of Inte-
gers as an argument and returns a new sequence of Integers. Inside the function, a for 
loop multiplies each sequence element by the scaling factor. The variable thisSeq  is 
necessary because JavaFX does not allow you to modify function parameters (seq ).

We could also use the mult  function to scale our original sequence.

var seq = [1, 2, 3, 4, 5]; // [ 1, 2, 3, 4, 5 ]
var multTen = mult(10); // scaling factor is 10

4. Specifically, a function literal implements a closure because it retains references to variables 
that are outside the function•s scope (i.e., variables that appear free in the closure).

seq = multTen(seq); // generate new sequence, replace old one
println(seq);           // [ 10, 20, 30, 40, 50 ]

Remember, the mult  function generates new sequences. Instead of changing the origi-
nal sequence, you are reassigning a new sequence to the original sequence variable.

A common use of function types is with event handlers for graphical objects. JavaFX 
nodes, for instance, let you specify function types for properties that describe event 
handlers. The following example shows a Rectangle component with an event han-
dler that acknowledges when a user clicks inside the Rectangle with the mouse.

Rectangle {
x: 40
y: 40
width: 40
height: 40
fill: Color.RED
onMouseClicked: function(e: MouseEvent): Void {

println("You clicked me");
}

}

Here, property onMouseClicked  has function type function(e: MouseEvent): Void .

3.6  Classes and Objects

Besides built-in types, JavaFX lets you create your own data types in applications. 
This powerful feature is the concept behind object-oriented design techniques that 
work well in complex •real worldŽ applications. In this section, we show you how to 
define your own classes and create objects in JavaFX. You•ll also learn how to use 
inheritance to create new classes from existing ones.

Classes

Let•s begin with a simple Point class.5

public class Point {
public var x: Number;

    public var y: Number;
    public function clear(): Void {
        x = 0;
        y = 0;

5. The Point class is in Point.fx  and the code that creates and uses Point objects is in Main.fx .
See •Access ModifiersŽ on page 88 for a more detailed discussion of how to use public  and 
other access modifiers with classes.

    }
}

The Point class manages x and y coordinates (both Number types). Every Point object 
has its own x and y (these are called instance variables). Since the instance variables are 
both Number types, they have default values of 0.0. The clear  function sets the x and y 
instance variables to zero. Note that clear  is defined inside the Point class. This is 
called an instance function. As you will see shortly, instance functions are always called 
with object references. This implies that the instance variables of the object are always 
in the scope of the instance function body. The class, instance variables, and instance 
function are all declared public so that code in separate files may access them.

Object Literals

Most object-oriented languages use special class functions called constructors to cre-
ate objects from classes. In JavaFX, you build an object from a class with an object lit-
eral expression. Here are several examples with the Point class.

// object literal initializes instance variables
var p = Point { x: 10, y: 20 };
println("p = ({p.x}, {p.y})");     // p = (10.0, 20.0)

// empty object literal uses defaults for instance variables
var q = Point {};
println("q = ({q.x}, {q.y})"); // q = (0.0, 0.0)

Object literals begin with a class name followed by curly braces "{}"  containing ini-
tializers with :  separating instance variables from their values. If you assign an object 
literal to a variable (object reference) in your program, you can call instance functions 
later with the object reference. An empty object literal uses defaults for the instance 
variables. JavaFX lets you use commas, semicolons, or newlines to separate initializ-
ers. Here we use an object literal on one line, but you can use separate lines for each 
initializer if you want.

Programming Tip

In addition to instance variable initializers, you can also declare local variables and define 
functions in object literals. Here•s an example that sets the x and y instance variables in a 
Point object literal using the return value of a function.

var r = Point { 
var n = getNumber()

    x: n
    y: n + 10

}

There are lots of advantages to creating objects with object literal expressions. First of 
all, you see the values that each instance variable is initialized to at the point where 
you construct the object. Second, object literals nest well, so you can use an object lit-
eral inside other object literals. Third, sequences of object literals may appear inside 
object literals. Fourth, you can construct object graphs without creating unnecessary 
temporary variables. All this makes object literals useful in the design of complicated 
things (graphical user interfaces, for instance), as you will see later on.

Recall that the Point class has a clear  instance function. When you call this function 
with a Point object reference, it resets that object•s x and y coordinates to zero.

p.clear();
println("p = ({p.x}, {p.y})");     // p = (0.0, 0.0)

Using public-init

Once you define a Point object, moving to a new spot is straightforward.

var p = Point { x: 10, y: 20 };
p.x = 30; p.y = 40; // move to (30, 40)

But there is a potential problem. Suppose our Point objects can only have positive val-
ues for their x and y coordinates. With our current class definition, an application 
could do the following.

var p = Point { x: 10, y: 20 };
p.x = -1; // change x coordinate to -1
p.y = -1; // change y coordinate to -1

This compiles and runs, but it•s not the behavior we want. We need a Point class with 
a more controlled interface. Let•s restrict the accessibility of instance variables x and y
and control how users move Point objects.

public class Point {
public-init var x: Number;

    public-init var y: Number;
init {

if (x < 0) x = 0;
if (y < 0) y = 0;

    }
    public function move(newX: Number, newY: Number): Void {

if (newX >= 0 and newY >= 0) {
x = newX;
y = newY;

}
    }
    public function clear(): Void {

x = 0;
y = 0;

    }
}

To change the location of point objects, the Point class has a new instance function 
called move. Note that this function won•t move to a different spot unless the new x

and y values are legal (not negative). But what•s to prevent users from changing x and 
y directly like before? The answer is a public-init access modifier with the declarations 
of instance variables x and y.

With instance variables, the public-init  access modifier allows an application to ini-
tialize its value in an object literal, but the instance variable is read-only after that.

Init Blocks

The other new feature introduced in this Point class is an init block.

init {
if (x < 0) x = 0;
if (y < 0) y = 0;

}

Init blocks are optional in classes, but if you write one and create an object with an 
object literal, the init block is executed immediately after the instance variables receive 
their values. 

We don•t want users to set x and y to negative values in object literals, so we use an init 
block to set them to zero if they are negative. Init blocks help you implement tasks 
during initialization and mimic things you typically do with constructors. JavaFX also 
has postinit blocks (see page 78).

These enhancements implement the behaviors we want.

var p = Point { x: 10, y: 20 }; // (10.0, 20.0)
p.move(30, 40); // move to (30.0, 40.0)
p.move(-10, -10); // does not move
p.x = -1; // compiler error
p.y = -1; // compiler error
var q = Point { x: -10, y: -20 }; // (0.0, 0.0)

Using public-read

Next, let•s suppose Point objects must maintain their distance from the origin (0,0 ).
Here•s a version of Point that implements this new feature.

import java.lang.Math;

public class Point {
public-init var x: Number;

    public-init var y: Number;
    public-read def distance = bind Math.sqrt(x * x  + y * y);

init {
if (x < 0) x = 0;

if (y < 0) y = 0;
    }
    public function move(newX: Number, newY: Number): Void {

if (newX >= 0 and newY >= 0) {
x = newX;
y = newY;

}
}
public function clear(): Void {

x = 0;
y = 0;

}
}

The Point class now has a def  instance variable called distance . The keyword public-

read  makes distance read-only when accessed by applications. 

var p = Point { x: 10, y: 20 }; // (10.0, 20.0)
println("length of p = {p.distance}");    // length of p = 22.36068
p.move(30, 40); // move to (30.0, 40.0)
println("length of p = {p.distance}");    // length of p = 50.0

However, distance  cannot be changed or used in an object literal.

p.distance = 50; // compiler error
var q = Point { x: 3, y: 4, distance: 5 }; // compiler error

The distance  variable also receives its value from the Java Math.sqrt  function. Note 
that we use bind  with distance  and Math.sqrt . This makes distance  update automati-
cally when x changes or y changes. Without bind , Math.sqrt  would be called only once. 
Binding is critical here, since it keeps distance  up to date as you create, move, or clear 
Point objects. 

Properties

Instance variables that are public , public-init , or public-read  are exposed to code outside 
the class. These specialized instance variables are called properties.

Using this

The move instance function for class Point can also be written this way.

public function move(x: Number, y: Number): Void {
if (x >= 0 and y >= 0) {

this.x = x;
this.y = y;

}
}

Here, the function•s parameters have the same names (x, y) as the instance variables in 
our Point class. Consequently, we need the keyword this  to distinguish between 
them. The expression this.x  refers to the instance variable in the class and x refers to 
the local function parameter. 

Using null

Recall that all JavaFX built-in types have default values. What about defaults for 
object references? If you don•t initialize or assign a reference to an object you create, its 
default is null . You can also assign null  to object references.

var p: Point; // p is null (default)
p = Point { x: 10, y: 20 }; // p refers to Point(10.0, 20.0)
p = null; // p is null

A common use for null  is to check the validity of a function•s argument when it is a 
class type.

function calculate(p: Point) {
if (p == null) {

/* object not initialized */
}

}

JavaFX Tip

JavaFX does not throw exceptions for null references. If you call an instance function with a 
null reference, the call is ignored. If you access an instance variable with a null reference, 
default values are used.

Using Java Objects

If you need to create Java objects in your JavaFX programs, use operator new and pass 
arguments directly to the Java class constructor, if necessary. To illustrate, here•s an 
example that uses the Java Date and DateFormat objects to give you the current time.

var today = new java.util.Date();
var timeNow = java.text.DateFormat.getTimeInstance().format(today);
println(timeNow);     // 4:41:30 PM

A new Java Date object is passed to getTimeInstance().format  and called with 
java.text.DateFormat  to return the current time as a string.

Here•s an example that uses Java character strings. Suppose you have an input string 
containing leading spaces and fields of substrings delimited by colons. You want to 
strip leading spaces and replace the colon characters in the input string with spaces. 

JavaFX strings are immutable, so you can•t modify the input string directly. But you 
can call Java functions trim  and replace  with JavaFX strings, as follows.

var str = "  Ted Smith:123 Elm St.:10105";
str = str.trim(); // trim leading spaces
def space: Character = 32; // "space" character
def colon: Character = 58; // ":" character
str = str.<<replace>>(colon, space);
println(str); // Ted Smith 123 Elm St. 10105

Sincereplace  is a JavaFX keyword, use <<replace>>  to call the Java string function 
replace . Note that both functions return new strings, which we use to update the orig-
inals.

Alternatively, you can use a Java StringBuffer object inside a JavaFX replaceSpaces

function. Here•s the code.

function replaceSpaces(input: String): String {
def space: Character = 32;
def colon: Character = 58;
var str = new java.lang.StringBuffer(input); // StringBuffer
var index = 0;
// strip any initial spaces first
while ((index <= str.length()-1) and (str.charAt(index) == space))

str.deleteCharAt(index);
for (i in [0..str.length()-1])

if (str.charAt(i) == colon)
str.setCharAt(i, space); // replace colon with space

return str.toString(); // return as String
}

var newString = replaceSpaces("  Ted Smith:123 Elm St.:10105");
println(newString);     // Ted Smith 123 Elm St. 10105

Inside function replaceSpaces , variable str  is a StringBuffer object created from the 
Java StringBuffer class using string input  as the argument to its constructor. The while 
loop calls StringBuffer methods charAt  and deleteCharAt  to find and delete leading 
spaces. The for loop uses StringBuffer method setCharAt  to replace colon characters 
with space characters. 

Binding with Object Literals

In JavaFX, you can bind directly to an object literal or bind any subset of its initializ-
ers. Binding directly to an object literal creates a new instance.

def v = bind ClassName  {
w: expression ;

}

If expression  changes, a new instance of ClassName  is created and assigned to v with w

having expression •s value.

Binding to an initializer in an object literal does not create a new instance.

def v = ClassName  {
w: bind expression ;

}

If expression  changes, w is updated from expression •s value in the same object.

Binding initializers in object literals is most useful with component properties. Here•s 
an example that loads images in a web service call (only skeletal code is shown here).

var loadComplete = false;
var description: String;

function makeServiceCall() {
description = "Loading Photos...";
. . .
loadComplete = true;

}

function alert(msg: String): Void {
loadComplete = false;
description = "Loading Error: {msg}";

}

scene: Scene {
content: [

VBox {
visible: bind not loadComplete // bind to visible property
content: [

Text {
content: bind description // bind to Text content

}
]

}
]

}

Two component properties are bound here. The visible  property of a VBox is bound 
to the loadComplete  boolean and the content  property of a Text component is bound to 
a description  string. Both bind expressions update the scene graph dynamically.

A call to makeServiceCall  starts loading images. The loadComplete  boolean is initially 
false, so the VBox is visible and its Text component displays "Loading Photos" . When 
image loading has completed, makeServiceCall  sets loadComplete  to true and the VBox 
becomes invisible. If an error occurs, the alert  function sets loadComplete  to false and 

changes the description  string to an error message. This makes the visible Text com-
ponent now display the error message.

Programming Tip

Binding component properties is a powerful JavaFX idiom that lets you update your scene 
graph dynamically.

Overriding bind

You can bind class instance variables with the keyword var . This lets you override the 
bind with an object literal initialization. Here•s an example.

var size = 100;

public class Thing {
var num = bind size; // num binds to size

}

var t1 = Thing { }; // t1.num is 100
size = 500; // t1.num is 500
var t2 = Thing { num: 50 }; // t2.num is 50
size = 300; // t1.num is 300, t2.num is 50

Sincenum is directly initialized in the object literal for t2 , the binding of size  does not 
apply to that object.

3.7  Inheritance

Code reuse is a major goal of object-oriented programming. When designing a new 
class, you can derive it from an existing one. This is called inheritance and represents 
an •is-aŽ relationship between classes. Inheritance makes it easy to hook into existing 
frameworks and use existing APIs. With inheritance, you can retain the state and 
behavior of an existing class and specialize certain aspects of it to suit your needs.

In JavaFX, inheritance is implemented by extending classes (just like Java). When you 
use the extends  keyword to inherit one class from another, the public functions of the 
•parentŽ class become part of the public functions of the •childŽ class. The parent 
class is often called a superclass and the child class a subclass.

JavaFX Tip

JavaFX classes can extend at most one Java or JavaFX class, and any number of Java interfaces 
or JavaFX mixins (see •Mixin InheritanceŽ on page 81).

Overriding Functions

Even if you don•t use the extends  keyword explicitly when you create a new class, the 
JavaFX compiler implicitly extends your new class from superclass Object. Let•s 
return to our Point class to see what this means.

public class Point extends Object  {
public-init var x: Number;
public-init var y: Number;
. . .
public override function toString(): String {

return "({x}, {y})";
}

}

The Object superclass includes an instance function called toString  that returns a 
String type for every object in JavaFX. Using the override  keyword, you can provide a 
different implementation for your extended class. In our Point class, for instance, the 
toString  function now returns the x and y coordinates as an ordered pair (x, y) . This 
means you can use a Point object in any expression that expects a String type and 
you•ll get a nicely formatted ordered pair.

var p = Point { x: 10, y: 20 }; // (10.0, 20.0)
println("p = {p}"); // p = (10.0, 20.0)
var spot = "{p}"; // spot is a String
println("spot = {spot}"); // spot = (10.0, 20.0)

You•ll see more examples of overriding functions in the next section.

Programming Tip

The default conversion for class types to Strings is the address of the object. To display an 
object•s state in a more meaningful way, override the toString()  function.

Using super

Suppose a Pixel class needs to manage color with x and y coordinates. You could 
design this class from scratch, but it would be a lot of work. Why not reuse the Point 
class and extend Pixel from Point? A Pixel, after all, is a Point object with color.

Here is the Pixel class.6 As you see, there is not a lot of code to write when you use 
inheritance.

import javafx.scene.paint.Color;

6. The Pixel class is in Pixel.fx  and the Point class is in Point.fx .

public class Pixel extends Point {
public var color: Color;
public override function toString(): String {

return "({color}, {x}, {y})";
}
public override function clear(): Void {

super.clear();      // call clear() in Point class
color = null;

}
}

A Pixel class manages a public color  instance variable. Its type (Color ) is accessible via 
the import statement shown. When a class defines an instance variable with public ,
applications have read and write access to that variable. (Contrast this with public-

init  in the Point class for x and y, which grants only read access of these instance vari-
ables to applications after object literal initialization). 

The Pixel class also overrides toString  to convert Pixel objects to the string "(color,

x, y) ". Clearing a Pixel object is also desirable. The behavior is to reset the x, y coordi-
nates to zero and make the Pixel object have no color (null ). To accomplish this, you 
need a way to call the clear  instance function in the Point class. That•s what 
super.clear()  does. Without this super  keyword, clear  would call itself recursively 
and the Pixel•s x and y coordinates would never change.

Let•s create several Pixel objects now and try them out.

var p = Pixel { color: Color.RED, x: 30, y: 40 };
println("p = {p}"); // p = ([red=255,green=0,blue=0,...], 30.0, 40.0)
println("length of p = {p.distance}");    // length of p = 50.0

var q = Pixel {}; // default values (color = null)
q.color = Color.BLUE; // change color
q.move(10, 10); // move to (10, 10)
println("q = {q}"); // q = ([red=0,green=0,blue=255,..], 10.0, 10.0)
println("length of q = {q.distance}");    // length of q = 14.142136

p.clear(); // clear Pixel
println("p = {p}");       // p = (null, 0.0, 0.0)
println("length of p = {p.distance}"); // length of p = 0.0

The first Pixel object (p) has a red color at position (30.0, 40.0) . The second Pixel 
object (q) is at the origin (0, 0)  with no color initially, but we change its color to blue 
and move it to (10.0, 10.0).  When we clear Pixel object p, its color is set to null and 
its x, y values are set to zero. Note that distance  (defined in class Point) is inherited 
and accessible with Pixel objects.

From these examples, you see that a Pixel is a Point with public-init  properties for x,
y, and read-only for distance . What makes a Pixel different from a Point is its color

property, which you can modify. Like Points, Pixels may be moved or cleared.

PostInit Blocks

JavaFX supports init blocks and postinit blocks inside classes. An init block runs 
immediately after the instance variables have been initialized. A postinit block exe-
cutes right after an object has been completely initialized. Here is an example with 
inheritance that shows you both.

class Superclass {
var m: Integer;
init { println("Superclass.init"); }
postinit { println("Superclass.postinit"); }

}
class Subclass extends Superclass {

var n: Integer;
init { println("Subclass.init"); }
postinit { println("Subclass.postinit"); }

}
var s = Subclass { m: 2, n: 3 };

The output shows init blocks in both the superclass and subclass always execute 
before any postinit blocks, even postinit blocks in the superclass.

Superclass.init
Subclass.init
Superclass.postinit
Subclass.postinit

JavaFX Tip

Postinit blocks are rarely used. Their primary purpose is to register listeners or otherwise pub-
lish a completely initialized object.

Abstract Base Classes

Another concept in object-oriented programming is a class that you can•t instantiate. 
These classes define abstract functions that you cannot invoke directly and exist only 
to specify contracts for state and behavior. Abstract functions must be overridden in 
subclasses.

If a class defines abstract functions (and possibly members), it•s called an abstract class.
An interface contains only abstract functions. JavaFX does not support interfaces with 
the keyword interface  like Java does,7 but it does have abstract base classes.

7. Java treats abstract classes as separate from interfaces, even if the abstract class has only 
abstract methods.

In JavaFX, the keyword abstract  defines abstract classes. The abstract  keyword also 
defines a function with no code. Here•s an abstract class called Shape2d that defines a 
two-dimensional shape with an area  function.

public abstract class Shape2d { // abstract class
public abstract function area(): Number; // abstract function

}

The Shape2d class does not have state, so no instance variables are necessary. Since 
the specific type of Shape2d is unknown at this point, area  must be abstract with no 
code. This class exists only to be inherited from, so let•s define two simple shapes, a 
Circle and a Square.

import java.lang.Math;

public class Circle extends Shape2d {
public var radius: Number;
public override function area(): Number { // redefine area

Math.PI * radius * radius;
}
public override function toString(): String {// redefine toString

return "Circle";
}

}

public class Square extends Shape2d {
public var side: Number;
public override function area(): Number { // redefine area

side * side;
}
public override function toString(): String {// redefine toString

return "Square";
}

}

You must override function area  in both Circle and Square (and any other class that 
extends Shape2d) because it•s defined as abstract in Shape2d. If you do not, compila-
tion errors occur. However, overriding toString  in both classes is not required (this 
function has default code in class Object and is not abstract). Here, toString  is over-
ridden to return the name of the class as a string.

Let•s create two shape objects and try them out.

var shapes: Shape2d[] = [
Circle { radius: 10 },

    Square { side: 20 }
];

for (s in shapes) {
println("{s} has area {s.area()}");

}

Variable shapes  is a sequence containing Circle and Square objects of type Shape2d. 
The for loop fetches each object from the sequence and prints their Shape names and 
areas. Here•s the output.

Circle has area 314.15927
Square has area 400.0

JavaFX Tip

When you create a custom graphical node in JavaFX, you extend CustomNode. You must over-
ride function create  because CustomNode defines abstract function create.

As Operator

Earlier in the chapter, we introduced the Boolean instanceof  operator (see •Instanceof 
OperatorŽ on page 50), which returns true if its argument is a specific type. JavaFX 
also includes an as  operator that returns a subclass type if its argument is an object of 
that type at runtime. With Object types, the instanceof  and as operators let you spe-
cialize tasks for specific subclass types. 

To illustrate, suppose you need to access the width of a Rectangle component in an 
event handler.

onMouseClicked: function(e: MouseEvent): Void {
println(e.node.width); // does not compile

}

The println  statement does not compile because e.node  is not a Rectangle type. To 
make this work correctly, you need the instanceof  and as  operators.

onMouseClicked: function(e: MouseEvent): Void {
if (e.node instanceof Rectangle) {

var w = (e.node as Rectangle).width;
println("Rectangle is {w} wide");

}
}

The instanceof  operator verifies that expression e.node  is a Rectangle before the as

operator converts it to a Rectangle type. Now you can access the width of the Rectan-
gle and print it with println .

Programming Tip

It•s a good idea to use the instanceof  and as  operators together to check for specific types at 
runtime. Without instanceof , the as  operator fails and throws a ClassCastException  if you 
give it the wrong type.

Mixin Inheritance

Occasionally you•ll want to combine behaviors and state from more than one class. 
This is called mixin inheritance. Using the mixin  keyword, you can create a mixin class.
Recall that a JavaFX class can extend at most one Java or JavaFX class and any number 
of Java interfaces. A JavaFX class can also extend any number of JavaFX mixin classes.

The format for a mixin class is

mixin class MixinClass  { . . . }

Mixin classes cannot be instantiated, just like abstract classes. A class that extends a 
mixin class with the keyword extends  is called a mixee.

class Mixee  extends MixinClass  { . . . } 

A mixin class is a type, so you can use its name with the instanceof  and as  operators.

Mixin classes can have public abstract functions that mixees must override, which is 
similar to abstract base classes and Java interfaces. If a mixin class has a non-abstract 
function, the mixee has the option of overriding the function in a subclass or using the 
default implementation from the mixin class.

Here•s an example that overrides functions with mixins.

public mixin class Mixin {
public function write(obj: Object): Void {

println("Mixin write");
}
public abstract function log(entry: String): Void; // no implementation

}

public class Base { . . . }

public class Derived extends Base, Mixin {
public override function write(obj: Object): Void {

println("Derived write");
}
public override function log(entry: String): Void { // must override

println("Derived log");
}

}

var d = Derived { };
d.write("something");    // Derived write
d.log("Derived");    // Derived log

The Mixin class defines a write  function with a default implementation and an 
abstract log  function. A mixee must implement a version of log  but is not required to 
provide an implementation for write . Here, the mixee (Derived) uses override  to 

implement its own version of write . A call to d.write("something")  invokes Derived•s 
write  function but if Derived did not override write , the one in the Mixin class would 
be called.

Programming Tip

You must use the keyword mixin  here. When you extend from more than one JavaFX class, 
only one non-mixin based class is allowed.

With mixin classes, you can resolve ambiguities in overridden functions with class 
names. To illustrate, look at the following classes.

public mixin class Log {
public function write(str: String): Void {

println("Log write");
}

}

public class Document {
public function write(str: String): Void {

println("Document write");
}

}

public class LogFile extends Document, Log {
public override function write(str: String): Void {

Log.write(str);
}

}
LogFile{}.write("something");      // Log write

Class LogFile extends from Document and a Log mixin class. The Log and Document 
classes both define write  functions with the same signature. LogFile overrides the 
write  function to call Log.write(str)  in the Log class. A call to Document.write(str)

would invoke the write  function in Document. Note that compilation errors occur 
with LogFile{}.write("something")  if LogFile does not override the write  function.

Mixin classes can also declare variables. A variable declaration has a name, type, 
optional default value, and possibly an on replace  trigger. A mixin with variable x

means mixees must also have an x (an x is provided if the mixee does not already have 
one; there is only one x in any case). Just like abstract functions, mixees have the 
option of using defaults for the mixin variable or extending the definition with the 
override  keyword. This allows the mixee to override the default value and/or add a 
different on replace  trigger. Triggers work with mixin classes just as they do with 
ordinary classes. (See •On Replace with VariablesŽ on page 83.)

Here•s an example that overrides variables with mixins.

public mixin class Mixin {
public var value: Number = 1.1;
public function getValue(): Number {

return value;
}

}

public class Base { . . .}

public class Derived extends Base, Mixin {
public override var value = 2.2;

}

var d = Derived { };
println(d.getValue());              // 2.2

The Mixin class defines a Number variable that the Derived mixee class overrides with a 
new value. In Derived, a type is not required for value , since Number is inferred. A call 
to d.getValue()  returns the overridden value ( 2.2 ). If Derived does not override value ,
d.getValue()  returns the Mixin value ( 1.1 ).

JavaFX Tip

Mixin inheritance is not the same as multiple inheritance. Unlike multiple inheritance, which 
can inherit variables with the same name and type from multiple superclasses, mixin inherit-
ance guarantees that only one variable of that name and type will be •mixedŽ in. This resolves 
ambiguities and makes mixin classes easier to work with.

3.8  Triggers

JavaFX has another feature that lets you execute code when variables in your program 
change their values. This is called a trigger and is implemented with the on replace

keywords. Triggers are similar to binding block expressions except that you are not 
limited to def  and var  statements inside your blocks (see binding block expressions on 
page 53). In JavaFX, you can use triggers with variables, properties, sequences, and 
with bind . Let•s look at several examples and show you how triggers work.

On Replace with Variables

The simplest use of on replace  is to react to a change in the value of a variable in your 
program. The format is

var v = value  on replace oldvalue  {
. . .

}

When variable v changes, the statements in the block execute. The oldvalue  name is 
optional and is set to the previous value of v. The block also executes when you initial-
ize variable v the first time.

Here•s an example of on replace  that only inserts positive values in a sequence.

var seq: Integer[];
var value = 0 on replace {

if (value > 0) insert value into seq;
}

value = 12; value = -5; // no negatives in seq
println(seq);      // [ 12 ]
value = 20;
println(seq);     // [ 12, 20 ]

When value  changes, the on replace  block executes. The if  statement makes sure that 
value  is positive before inserting it into the sequence.

If you need the previous value when an on replace  block executes, include a name 
after the keyword replace  (any non-keyword name will do). Here•s an example that 
shows you how this works.

var number = 4 on replace old {
println("old = {old} new = {number}");   // old = 0 new = 4

}

number = 8;     // // old = 4 new = 8

Note that println  executes twice here, once when number  receives its initial value (4) 
and again when number  changes to 8. The old  variable is 0 the first time because that•s 
the default value for number .

On Replace with Sequences

You can use on replace  with sequences, too. The format is

var seq  = [ sequence ] on replace oldValue [ low .. high ] = newSeq {
. . .

}

When sequence seq  changes, oldValue  is the previous sequence, low  and high  are the 
indices of the sequence that changed, and newSeq is the new sequence. (You can use 
any non-keyword for these names.) Note that oldValue [ low .. high ]  is a sequence slice.

Here•s an example to see how all this works.

var chars = ['a','b','c'] on replace old[lb..ub] = newchars {
println("old = {old} low = {lb} high = {ub} newchars = {newchars}");

}         // old =  low = 0 high = -1 newchars = abc

insert 'd' into chars; // old = abc low = 3 high = 2 newchars = d
delete 'a' from chars; // old = abcd low = 0 high = 0 newchars =
chars[1..2] = ['i', 't']; // old = bcd low = 1 high = 2 newchars = it
println(chars); // [ b, i, t ]

When the chars  sequence is initialized, old  is empty and newchars  is the new 
sequence. Because this is an insertion, low  is 0 and high  is -1 (one less than low ). The 
other sequence expressions generate the values shown (left as an exercise to the 
reader). Note that newchars  is empty when you delete from the sequence.

On Replace with isInitialized

The isInitialized  function returns true if its argument has been previously set. This 
can be important in on replace  blocks, since they always execute at least once. If a vari-
able in your program receives a default value, you may not want to execute the state-
ments in an on replace  block. The isInitialized  function can prevent this.

To illustrate, here•s a Distance object that converts miles to kilometers and vice versa.

class Distance {
def factor = 1.609344;
var miles: Number on replace {

if (isInitialized(miles))
kilometers = miles * factor;

}
var kilometers: Number on replace {

if (isInitialized(kilometers))
miles = kilometers * 1.0 / factor;

    }
function print() {

println("Miles = {%6.3f miles}, Kilometers = {%6.3f kilometers}")
}

}

Distance{miles: 60}.print();        // Miles = 60.000, Kilometers = 96.561
Distance{kilometers: 500}.print();  // Miles = 310.686, Kilometers = 500.000

Programming Tip

Inside both on replace  blocks, isInitialized  verifies that variables miles  and kilometers

are set before they are used. This avoids executing the conversions for this situation.

Distance{}.print();        // Miles = 0.000, Kilometers = 0.000

On Replace with Bind

Another nice technique with on replace  is to use it with bind . The format is

def v = bind w on replace oldValue  {
. . .

}

Here, if w changes, v is updated to w•s  value and then the code in the on replace  block 
executes. As before, oldValue  is optional.

Programming Tip

Using def  gives you compiler errors if you modify v. If you use var , you•ll get an AssignTo-

BoundException  at runtime. In general, use def  in declarations with bind.  (The exception 
occurs with class instance variables when users can override binding with an object literal. See 
•Using thisŽ on page 71.)

To see an example of on replace  with bind , let•s return to our image loading example 
from earlier (see page 74). Recall that a loadComplete  boolean was set to true by func-
tion makeServiceCall  when image loading was completed. Here is a code snippet that 
binds a loadCarousel  variable to the loadComplete  boolean and starts up the carousel 
when all the images are loaded.

def loadCarousel: Boolean = bind loadComplete on replace {
if (loadComplete) {

println("starting carousel");
carousel.play();

}
}

Programming Tip

Remember, the on replace  block executes when you first initialize the loadComplete  boolean 
to false. Here we check the flag and make sure it•s true before starting up the carousel.

3.9  Script Files and Packages

This section discusses program structure, how to access variables and functions from 
different script files, and how to organize your code into packages.

JavaFX lets you structure code as a collection of script level statements (statements 
that are not inside a function or class) or as reusable packages. Let•s look at these two 
approaches in more detail and show you how JavaFX keywords help control the 
accessibility of your variables, objects, and functions.

Variable Scope

There are three types of variables in JavaFX: script variables, class instance variables, 
and local variables. Script variable declarations appear at the top-level of a JavaFX 
script outside of function definitions. Their values are visible everywhere in the script 
file (and, if public , everywhere in the program). The lifetime of a script variable starts 
when the script is loaded and ends when the program terminates. Local variable dec-
larations only appear inside blocks (curly braces); their lifetime ends when you exit the 
block. Class instance variables appear inside JavaFX class definitions. Their scope is 
tied to the lifetime of the object they belong to.

Here is an example of variable declarations and their scope.

var interval = 100ms; // script variable

class Thing {
var val: Integer; // instance variable

}

function doSomething(t: Thing): Void {
var timeSlice = t.val * interval; // local variable
simulate(timeSlice);

}

Function Scope

JavaFX lets you write script functions and class instance functions. A script function is 
callable from anywhere in the script file where it•s defined. Instance functions, on the 
other hand, are defined inside class declarations and must be called with object refer-
ences.

Here are examples of script functions and instance functions.

function totalSum(seq: Integer[]) { // script function
var sum;
for (i in seq) sum += i;
return sum;

}

class Square {
var side: Number;
function area(): Number { side * side; } // instance function

}

As you will see shortly, you can control the accessibility of your JavaFX variables and 
functions with access modifiers.

Script Files

If you want to write JavaFX code quickly and check out its behavior, JavaFX lets you 
organize a program into script files. With this approach, it•s not necessary to include a 
run  function or use packages. This can be very useful when you are learning or just 
want to check out a portion of JavaFX code.

To illustrate, here•s a script file that defines a Complex class and an add  function to 
add Complex objects.

class Complex {
var real: Number;
var imag: Number;
override function toString(): String {

return "({real}, {imag})";
}

}

function add(a: Complex, b: Complex): Complex {
return Complex { real: a.real + b.real, imag: a.imag + b.imag; }

}

// script-level statements
var c1 = Complex{ real: 1.2, imag: 3.4 };
var c2 = Complex{ real: 3.1, imag: 2.5 };
var c3 = add(c1, c2);
println(c3);     // (4.3, 5.9)

Programming Tip

If you designate any function or variable public  in a script file, script-level statements will not 
compile. You have to put script-level statements in a run  function. A script file is a compilation 
unit in JavaFX.

Access Modifiers

To make JavaFX programs more modular and reusable, you can use access modifiers 
in your script files. Table 3.3 lists the primary access modifiers that apply to creating 
and using classes, calling functions, and reading and writing script variables or 

instance variables. These access modifiers also apply when overriding and setting or 
binding object literals of instance variables.

JavaFX Tip

JavaFX does not have a private access modifier like other languages (Java, C++). The default 
(no access modifier) is script-private.

Table 3.4 lists the variable access modifiers. You can use protected  and package  with 
public-read  and public-init  to grant additional write permissions as indicated in 
Table 3.4.

Packages

Let•s restructure the previous Complex script file and show you how to use it with 
packages. Here•s the approach.

// Complex.fx - Complex class

TABLE 3.3 Primary Access Modifiers

(no access modifier) Script-private. Accessible only within script where it is defined 
(default). Readable and writable.

package Only accessible from within package where it is defined. Readable 
and writable.

protected For class instance variables and functions, accessible to subclasses 
of the class where it is defined. For script variables and script 
functions, accessible from the same package. Readable and writ-
able.

public Accessible anywhere. Readable and writable.

TABLE 3.4 Variable Access Modifiers

public-read Writable only within the current script. Read-only outside the 
script.

public-init Writable only within the current script. Can be initialized in 
an object literal and read anywhere outside the script. Read-
only after initialization outside the script.

protected public-read Writable with sub-classes; otherwise read-only outside the 
script.

protected public-init Writable with sub-classes outside the script; otherwise init-
only.

package public-read Writable in the same package; otherwise read-only.

package public-init Writable in the same package; otherwise init-only.

package complex;

public class Complex {
public var real: Number;
public var imag: Number;
public override function toString(): String {

return "({real}, {imag})";
}

}

public function add(a: Complex, b: Complex): Complex {
return Complex { real: a.real + b.real, imag: a.imag + b.imag; }

}

The file Complex.fx  contains the code for class Complex and the add  script function. 
All members and functions have public  access specifiers and belong to package com-

plex .

Here•s a Main.fx  program that instantiates Complex objects and adds them.

// Main.fx - Complex objects
package complex;

var c1 = Complex { real: 1.2, imag: 3.4 };
var c2 = Complex { real: 3.1, imag: 2.5 };
var c3 = Complex.add(c1, c2);
println(c3);     // (4.3, 5.9)

The package  statement lets you access public members (Complex, add). Note that Com-

plex.add  is required to call the add  function. Access to public members is also possible 
with import statements.

Programming Tip

With functions, type inference can generally determine the types of your arguments and 
return values. However, you must explicitly provide types for functions that you intend to call 
from other script files.

JavaFX does not support static member variables and static member functions like 
other languages (Java, C++). You can, however, use script variables with appropriate 
access modifiers to simulate static behaviors. Here•s an example.

// Thing.fx - Thing class
package thing;

public def size1 = 100;
public-read var size2 = 200;

public class Thing {
public var num: Integer;

public function func() {
return size1 + size2;

}
}

public function myFunction() {
size1 = 300; // compiler error
size2 = 400; // ok

}

// Main.fx - Main program
package thing;

var th = Thing { num: 10 };
println(th.func());                // 300
println(thing.Thing.size2); // 200
thing.Thing.size2 = 500; // compiler error

In Thing.fx , size2  is modifiable but in other files ( Main.fx ) it is not. The notation 
thing.Thing  (package name and file name) is necessary to access script variables in 
Thing.fx . Note that public-read  is legal with script variables.

3.10  Exception Handling

In most cases, the JavaFX compiler does a good job of reporting errors at compile time 
when you do something wrong. Occasionally, however, unexpected runtime errors 
occur. In some situations, you might want to deal with runtime errors in some specific 
way. This is called exception handling. In this section, we show you how to handle 
exceptions in your JavaFX programs.

Try, Catch, Finally

Use the keywords try , catch , and finally  for thrown exceptions. The formats are

try {
// critical code

} catch (e: myException ) {
// exception handler code

}
finally {

// always executes
}

try {
// do something here
return;

}

finally {
// always executes

}

A try  block encloses code where an exception could be thrown. A catch  handler has a 
signature with an exception type ( myException ) and catches the thrown exception if 
the type matches. You can have more than one catch  handler (each with a different 
signature) and a handler can rethrow the same exception or a different one. A finally

block is optional and always executes, regardless of whether an exception was thrown 
or not. If you omit a catch  handler after a try  block, you must include a finally  block. 
The finally  block typically includes cleanup code.

Exceptions can easily be thrown in a JavaFX program, especially if you are using Java 
API methods. Here•s an example with a run  function that converts program argu-
ments.

// Numbers.fx - Program argument conversions
import java.lang.*;

function run(args: String[]) {
try {

def intVal = Integer.parseInt(args[0]);
def doubleVal = Double.parseDouble(args[1]);
println(intVal);
println(doubleVal);

} catch (e: NumberFormatException) {
println(e);

}
}

Without try  and catch , the program throws an exception at runtime and terminates if 
the program arguments have illegal characters for integers or doubles. To detect these 
runtime errors under program control, enclose the code in a try  block with a catch

handler that prints the error message.

$ javafx Numbers 123 45.67
123
45.67

$ javafx Numbers badchars
java.lang.NumberFormatException: For input string: "badchars"

Throwing Exceptions

The keyword throw  lets you throw exceptions in JavaFX. The format is

throw new myException ( args );

Note that you must use operator new to create an exception object from the myExcep-

tion  class. The constructor arguments typically contain information pertinent to the 
error. You can throw built-in Java exception objects or user-defined exception objects.

Here•s a checkArgument  function that throws an IllegalArgumentException  if its integer 
argument is negative.

import java.lang.*;

function checkArgument(num: Integer): Void {
if (num < 0)

throw new IllegalArgumentException("neg value");
}

JavaFX Tip

Unlike Java, JavaFX does not have throws clauses on script functions or instance functions 
that throw exceptions.

3.11  JavaFX Keywords

Table 3.5 lists the keywords and reserved words in JavaFX.

Use <<keyword >> to escape a keyword.

str = str.<<replace>>(colon, space); // call Java function replace

TABLE 3.5  JavaFX Keywords and Reserved Words

abstract after and as assert at

attribute before bind bound break catch

class continue def delete else exclusive

extends false finally first for from

function if import indexof in init

insert instanceof into inverse last lazy

mixin mod new not null on

or override package postinit private protected

public-init public public-read replace return reverse

sizeof static step super then this

throw trigger true try tween typeof

var where while with

4 Graphical Objects

Graphical objects and their manipulation are where JavaFX excels. This chapter shows 
you how to define and control graphical objects in JavaFX.

We begin at the top-level Stage and its Scene, followed by the all-important Node 
class. Nodes share a wealth of properties that let you manipulate graphical objects in a 
consistent way.

Nodes specialize into different types: there are nodes that are Shapes, nodes for 
grouping and layout, or developer•s can define their own nodes. Shapes come in all 
forms: Circle, Polygon, Line, and even Path„a shape that lets you describe your own 
Shape.

JavaFX graphical objects also include user interface components. We cover these in the 
next chapter. 

What You Will Learn

€ Top-level JavaFX objects Stage and Scene

€ Node class and its properties

€ Mouse and key event handler properties

€ Cursor types

€ Group and CustomNode

€ Shapes, Paths, and Path Elements

€ Layout components and bounding rectangles

4.1  Setting the Stage

JavaFX programs that render graphical material include a top-level Stage. The Stage 
includes a Scene object, which in turn includes a sequence of Nodes. By default the 
Stage style is set to StageStyle.DECORATED , which is rendered differently depending on 

the environment (Windows Vista, Mac OS, Windows XP). StageStyle.UNDECORATED

removes the decoration from the window. Both styles are shown in Figure 4.1.

JavaFX Tip

If you set the Stage•s style to StageStyle.UNDECORATED , you can provide your own •window 
closeŽ button. See the code in Figure 4.19 on page 116 for an example.

Scene

The JavaFX Scene class is the root for all content in a scene graph. The background of 
the scene is specified by the fill  property (default is Color.WHITE ). The sequence of 
Nodes in the content sequence is rendered on the scene.

Node

Class Node is the base class for all objects in the scene graph. You can add Node 
objects (subtypes of Node) to a scene graph, specify their properties, and apply trans-
formations. Node has many properties that let you customize its look and behavior, 
such as event handlers, clip , effect , opacity , rotate , properties for scaling and trans-
lating, visible , focusable , disable , bounding rectangles, and other properties listed in 
Table 4.1 and Table 4.2. 

Figure 4.2 shows a class hierarchy diagram of some of the JavaFX graphical objects 
discussed in this chapter. 

Stage {
title: "Stage Title"
width: 180
height: 150
style: StageStyle.UNDECORATED
scene: Scene {

fill: Color.BISQUE
content: // content here

}
}

Figure 4.1 Top-level Stage

Default Stage style StageStyle.UNDECORATED

Nodes have different subtypes, and as such, exhibit different specializations. For 
example, Group is a type of node that contains multiple subnodes in a content

sequence. Layout nodes are all types of Container, which is a kind of Group. (See 
•Layout ComponentsŽ on page 119 for a description of the layout nodes.) Shape is a 
type of subnode that is rendered in the coordinate space with properties such as fill
and stroke . Path, in turn, is a subtype of Shape that is made up of connected path ele-
ments. Circle is also a subtype of Shape. All these Node subtypes share common 
properties that help customize rendering, shown in Table 4.1. (See •Cascading Style 
Sheets (CSS)Ž on page 148 for details on applying styles to JavaFX nodes.) 

TABLE 4.1 Node Rendering Properties 

Property Type Description

clip Node Defines the clipping shape for this Node.

cursor Cursor Mouse cursor for this node and subnodes. If 
null, uses the first parent that has a non-null 
cursor. If still null, uses the cursor of the Scene.

effect Effect Specifies the effect to apply to this node. (See 
Table 8.3 on page 263 for a list of common 
JavaFX Effects.)

opacity Number Specifies how solid (opaque) a node appears (0 
is fully translucent and 1 is fully opaque). Note 
that opacity applies to the entire node, while 
the opacity argument with fill color affects only 
the fill (and not the stroke for example).

Figure 4.2 Node class hierarchy (partial)

Node

HBox

Group Shape

Path

Rectangle

Circle

Ellipse

. . .

clip, cursor, effect, . . .

fill, stroke, . . .content

CustomNode

Container

Flow

VBox

Stack

Tile

rotate Number Angle of rotation about the node•s center, mea-
sured in degrees. 

scaleX Number Factor by which coordinates are scaled about 
the center of the node along the x axis. -1 is nor-
mal scale, but the object is flipped.

scaleY Number Factor by which coordinates are scaled about 
the center of the node along the y axis. -1 is nor-
mal scale, but the object is flipped.

transforms Transform[] Defines sequence of Transform objects that 
apply to this node. Transformations are applied 
before translateX , translateY , scaleX , sca-
leY , and rotate . Transform objects include 
Translate, Scale, Rotate, or Shear.

translateX Number Defines x coordinate of the translation to apply 
to this node. Used to move a node with anima-
tion.

translateY Number Defines y coordinate of the translation to apply 
to this node. Used to move a node with anima-
tion.

layoutX Number Defines x coordinate layout adjustment to 
apply to this node. Used to position a node. 
(New in JavaFX 1.2.)

layoutY Number Defines y coordinate layout adjustment to 
apply to this node. Used to position a node. 
(New in JavaFX 1.2.)

visible Boolean If true, node should be rendered. If false, node 
is not rendered and will not receive mouse 
events or keyboard focus.

id String Similar to the •idŽ attribute of an HTML ele-
ment; useful to assign style elements from 
external style sheets.

style String Provides inline CSS styles. Uses the syntax 
defined in JavaFX CSS parser.

styleClass String Useful to assign style elements from external 
style sheets to all nodes matching styleClass .

TABLE 4.1 Node Rendering Properties (Continued)

Property Type Description

Programming Tip

If you•re dealing with movement or animation, use properties translateX  and translateY . If 
you•re dealing with node positioning, use properties layoutX  and layoutY .

Table 4.2 lists Node properties that let you customize how a node handles various 
input events (mouse and key events) and other properties that affect its event han-
dling.

TABLE 4.2 Node Mouse/Keyboard Event Properties 

Property Type Description

blocksMouse Boolean If true, consumes mouse events in node 
and does not send them to other nodes 
further up the scene graph. Commonly 
used when nodes overlap.

focused Boolean If true, node is current input focus owner. 
Only one node at a time may be the cur-
rent focus owner.

focusable Boolean If true, node can accept focus.

hover Boolean If true, node is currently being hovered 
over (typically with a mouse).

pressed Boolean If true, node is pressed (typically, the pri-
mary mouse button is down).

disable Boolean Sets disabled state of node (and any 
subnodes).

disabled Boolean If true, node is disabled. A disabled node 
should render itself differently.

onKeyPressed function(:KeyEvent)

:Void
Function called when node has input 
focus and a key has been pressed.

onKeyReleased function(:KeyEvent)
:Void

Function called when node has input 
focus and a key has been released.

onKeyTyped function(:KeyEvent)

:Void
Function called when node has input 
focus and a key has been typed (pressed 
and released).

onMouseClicked function(:

MouseEvent):Void
Function called when mouse button has 
been clicked (pressed and released).

onMouseDragged function(:

MouseEvent):Void
Function called when mouse button is 
pressed and then dragged.

onMouseEntered function(:

MouseEvent):Void
Function called when mouse enters node.

Key Events Tip

Use Node function requestFocus  to request focus for nodes that have key event handlers. 

Event handling for mouse and key events lets the user initiate actions. Let•s look at a 
simple mouse event handler that successively rotates a rectangle by 45 degrees, as 
shown in Figure 4.3. View A shows the default state of the rectangle. In View B, the 
user moves the mouse over the rectangle, causing the onMouseEntered  event handler to 
be called, which changes the rectangle•s fill color and stroke characteristics. When the 
user clicks the mouse, the rectangle rotates (View C). When the mouse exits the rect-
angle, the fill color and stroke are returned to the default settings. A double-click 
resets the rotation back to zero. 

Listing 4.1 shows the Rectangle object literal and its three mouse event handlers. 

Listing 4.1 Detecting Mouse Events

Rectangle {
    x: 30, y: 30

onMouseExited function(:

MouseEvent):Void
Function called when mouse exits node.

onMouseMoved function(:

MouseEvent):Void
Function called when mouse cursor 
moves within node but no buttons have 
been pushed.

onMousePressed function(:

MouseEvent):Void
Function called when mouse button has 
been pressed.

onMouseReleased function(:

MouseEvent):Void
Function called when mouse button has 
been released.

onMouseWheel-
Moved

function(:

MouseEvent):Void
Function called when mouse scroll wheel 
has moved.

TABLE 4.2 Node Mouse/Keyboard Event Properties (Continued)

Property Type Description

Figure 4.3 Detecting mouse events in a node

View A View B View C

    width: 90, height: 50
    stroke: Color.SLATEGRAY
    strokeWidth: 3
    fill: Color.CORNFLOWERBLUE
    onMouseClicked: function(e: MouseEvent): Void {
        if (e.clickCount == 2) { e.node.rotate = 0; }
        else  { e.node.rotate += 45; }
    }
    onMouseEntered: function(e: MouseEvent): Void {
        (e.node as Shape).fill = Color.BLUE;
        (e.node as Shape).strokeDashArray = [9, 5];
    }
    onMouseExited: function(e: MouseEvent): Void {
        (e.node as Shape).fill = Color.CORNFLOWERBLUE;
        (e.node as Shape).strokeDashArray = null;
    }
}

Note that expression (e.node as Shape)  is required before accessing Shape-specific 
properties fill  and strokeDashArray . (Expression (e.node as Rectangle)  also works 
for properties fill  and strokeDashArray .) The expression is not necessary with prop-
erty rotate , which is a Node property. You also don•t need instanceof  here, since the 
event handler object literal appears in a Rectangle, a subtype of Shape.

Cursor

JavaFX provides various cursor types, as shown in Figure 4.4 and listed in Table 4.3. 
You can set the cursor type of any node with property cursor , as follows.

Rectangle {
cursor: Cursor.HAND
width: 50
height: 100

}

The cursor type propagates to all subnodes unless a subnode redefines it. If cursor is 
null, the first parent with a non-null cursor defines the current node•s cursor type. If 
no cursor is defined within the scene graph, the cursor type of the Scene defines the 
cursor.

Figure 4.4 Cursor types

Left to right: crosshair, default, hand, move, text, wait, horizontal (east, west) resize, 
vertical (north, south) resize, northwest or southeast resize, northeast or southwest resize

There are also multiple cursor types for specialized resizing operations, as listed in 
Table 4.4.

Group

Group is a container class that groups nodes together. Most often, you use Group to 
construct a portion of a scene graph with related nodes. Once these nodes are inserted 
into the group•s content sequence, you can reposition the group without affecting the 
relative position of any of its subnodes. You build a Group object literal with sub-
nodes in property content . Alternatively, you can insert nodes under program control, 
as shown here.

var g1 = Group { }; // instantiate Group g1, empty content sequence
insert Rectangle {

width: 40
height: 20

TABLE 4.3 Common Cursor Types

Cursor Description

CROSSHAIR cross-shaped cursor

DEFAULT arrow

HAND hand

MOVE 4-direction arrow

TEXT I-bar

WAIT hour glass

TABLE 4.4 Resize Cursor Types 

Cursor Description

H_RESIZE double ended horizontal arrow

E_RESIZE

W_RESIZE

V_RESIZE double ended vertical arrow

N_RESIZE

S_RESIZE

NW_RESIZE double ended northwest to 
southeast arrowSE_RESIZE

NE_RESIZE double ended northeast to south-
west arrowSW_RESIZE

} into g1.content; // add a Rectangle to g1 content

CustomNode

Class CustomNode is an abstract base class that extends Node. It lets you create cus-
tom nodes for scene graphs. When you extend from CustomNode, you provide a cre-

ate  function. Function create  typically returns a Group node defining other nodes in 
its content  property. Here is the structure of the GuitarString custom node found in 
Chapter 2 (see Listing 2.1 on page 36 and Listing 2.2 on page 38).1

public class GuitarString extends CustomNode {
// properties, variables, functions
. . . 
protected override function create(): Node {
// put any initialization code here

return Group {
content: [

Rectangle { ... }
Rectangle { ... }
Rectangle { ... }
Text { ... }

]
} // Group

}
} // GuitarString

See •PianoKey ComponentsŽ on page 169 for another example that extends Custom-
Node.

4.2  Shapes

Class Shape is an abstract subclass of JavaFX Node (see Figure 4.2 on page 97). Shape 
objects inherit all properties in Node and adds new properties such as fill  and stroke

(outline drawn around the shape). Property fill  is type Paint, which can be a color, 
such as Color.RED , or a gradient (LinearGradient or RadialGradient). The default fill

value for all shapes is Color.BLACK  except Line, Polyline, and Path, which use default 
value null. You can make shapes transparent by setting their opacity  property to 0. 
You can alternatively provide an opacity argument with a color value or use 
Color.TRANSPARENT for property fill . (Property opacity  affects the entire node, includ-
ing stroke , whereas providing an opacity modifier with a fill color affects the fill

property only.) See •GradientsŽ on page 26 and •ColorŽ on page 28 for examples of 
gradients and color.

1. Function create  is called from the init  block in class CustomNode. If you need to do any 
initialization work prior to the custom node•s content being created, put that code in create .

Property stroke  is also type Paint. Shapes include several additional properties that 
let you configure the size, color, and look of a Shape•s outline stroke.

Table 4.5 lists a Shape•s properties.

Figure 4.5 shows three different line cap styles with property strokeLineCap . The top 
line uses StrokeLineCap.SQUARE  (the default), the second one uses StrokeLine-

Cap.ROUND, and the bottom line uses StrokeLineCap.BUTT .

TABLE 4.5  Shape Properties 

Property Type Description

fill Paint Interior of a Shape using the settings of the Paint 
context. The default value is Color.BLACK  for all 
shapes except Line, Polyline, and Path, which use 
default value null.

smooth Boolean If true , antialiasing rendering hints are applied 
when rendering the Shape.

stroke Paint Parameters of a stroke that is drawn around the 
outline of a Shape using the settings of the specified 
Paint. The default value is null for all shapes except 
Line, Polyline, and Path, which use default value 
Color.BLACK .

strokeDashArray Number[] A sequence representing the lengths of the dash 
segments. Alternate entries in the sequence repre-
sent the lengths of the opaque and transparent seg-
ments of the dashes. The pen is opaque when its 
current cumulative distance maps to an even ele-
ment of the dash sequence and transparent other-
wise.

strokeDashOffset Number Index in the dashing pattern that will correspond to 
the beginning of the stroke.

strokeLineCap Stroke-
LineCap

End cap style as one of the following values: 
StrokeLineCap.BUTT , StrokeLineCap.ROUND , and 
StrokeLineCap.SQUARE .

strokeLineJoin Stroke-
LineJoin

Decoration applied where path segments meet as 
one of the following values: StrokeLine-
Join.BEVEL , StrokeLineJoin.MITER , and Stroke-
LineJoin.ROUND

strokeMiterLimit Number Limit on ratio of miter length to stroke width for 
StrokeLineJoin.MITER  line join style.

strokeWidth Number Square pen line width. A value of 0.0 specifies a 
hairline stroke.

Listing 4.2 shows the object literals for each of the lines shown in Figure 4.5.

Listing 4.2 Property strokeLineCap

def line1 = Line {
    startX: 10, startY: 20, endX: 200, endY: 20
    strokeWidth: 8
    stroke: Color.BLACK
    strokeLineCap: StrokeLineCap.SQUARE
}
def line2 = Line {
    startX: 10, startY: 50, endX: 200, endY: 50
    strokeWidth: 8
    stroke: Color.BLACK
    strokeLineCap: StrokeLineCap.ROUND
}
def line3 = Line {
    startX: 10, startY: 80, endX: 200, endY: 80
    strokeWidth: 8
    stroke: Color.BLACK
    strokeLineCap: StrokeLineCap.BUTT
}

Figure 4.6 illustrates dashed lines, which are rendered with a sequence of numbers 
provided in property strokeDashArray . If you specify one number, a dashed line will 
have equal parts opaque (visible) and equal parts translucent. Otherwise, the num-
bers in the sequence alternate between the opaque part and the translucent part. The 
top line uses sequence [8 12] , which sets the visible dash size to 8 pixels and the 
translucent size to 12. The middle line uses [16 4] , to make the dash 16 and the trans-
lucent part 4. The bottom line uses 10 which provides equal-sized dashes and translu-
cent parts. In general, the wider the stroke (property strokeWidth ), the larger you 
should make your stroke dash sequence numbers.

Figure 4.5 Line cap styles 

SQUARE (default)

ROUND

BUTT

Listing 4.3 shows the object literals for each of the lines shown in Figure 4.6.

Listing 4.3 Property strokeDashArray

def line1 = Line {
    startX: 10, startY: 20, endX: 200, endY: 20
    strokeWidth: 2
    stroke: Color.BLACK
    strokeDashArray: [8,12]
}
def line2 = Line {
    startX: 10, startY: 50, endX: 200, endY: 50
    strokeWidth: 2
    stroke: Color.BLACK
    strokeDashArray: [16,4] 
}
def line3 = Line {
    startX: 10, startY: 80, endX: 200, endY: 80
    strokeWidth: 2
    stroke: Color.BLACK
    strokeDashArray: 10
}

It•s also possible to have more than two numbers in strokeDashArray . Even index num-
bers map to the opaque part of the dash and odd index numbers map to the translu-
cent part. This lets you build dashed lines with varying-sized dashes. For example, the 
following strokeDashArray  sequence creates a dashed line that cycles through pro-
gressively smaller dashes (16, 12, 8, 4 ) with equal spacing in between (8).

strokeDashArray: [16, 8, 12, 8, 8, 8, 4, 8]

When the number of elements in the strokeDashArray  is odd, numbers are assigned to 
opaque and translucent parts in order. In the following example, 20 is assigned to the 
opaque part in the first cycle and the translucent part in the second cycle.

strokeDashArray: [20, 4, 12]

Figure 4.6 Dashed line styles

strokeDashArray: [8,12] 

strokeDashArray: [16,4] 

strokeDashArray: 10 



Shapes 107

Rectangle

Rectangles have height  and width  properties and x and y properties for the upper left 
corner origin (default is 0 for the coordinate properties). For rounded rectangles, use 
properties arcHeight  and arcWidth  and specify the vertical and horizontal diameter of 
the arc at the four corners of the rectangle. Figure 4.7 shows the code for rendering the 
rounded Rectangle shown.

Book Examples

€ •Class WhiteKey„Part 2Ž, Listing 6.3 on page 172, (Chapter 6). Uses a rectangle to 
render a piano key.

€ •Class BlackKey„Part 2Ž, Listing 6.6 on page 177, (Chapter 6). Uses a rectangle to 
render a piano key.

€ •Photo1: Displaying an imageŽ, Listing 8.3 on page 254, (Chapter 8). Uses a rectan-
gle to frame an image.

Circle

A Circle has a center point (properties centerX  and centerY ) and property radius .
Figure 4.8 shows an example object literal that renders the Circle shown.

Rectangle {
    x: 30, y: 10
    width: 60, height: 80
    arcWidth: 20
    arcHeight: 20
    fill: Color.CORAL
    stroke: Color.BLUE
}

Figure 4.7 Rounded Rectangle example

Circle {
    centerX: 50, centerY: 50
    radius: 40
    fill: Color.CORAL
    stroke: Color.BLUE
}

Figure 4.8 Circle example

Book Examples

€ •Class WhiteKey„Part 2Ž, Listing 6.3 on page 172, (Chapter 6). Uses a circle to 
indicate a mouse event on a piano key.

€ •Using Timeline to animate a CircleŽ, Listing 7.1 on page 209, (Chapter 7). Ani-
mates a circle.

€ •PathBallŽ, Listing 7.21 on page 239, (Chapter 7). Creates a custom node with a cir-
cle.

Ellipse

Ellipse is similar to Circle, except that it has separate radius values for the x and y 
directions. Besides centerX  and centerY , you also specify radiusX  and radiusY .
Figure 4.9 shows an example object literal that renders the Ellipse shown.

Arc

Arcs have a center point (centerX  and centerY ) and two radii ( radiusX  and radiusY ),
similar to Ellipse. Arcs also have a starting angle (startAngle ) and an extent (length ).
Both startAngle  and length  are expressed in degrees. Arc objects have three types: 
ArcType.ROUND  (close the shape by connecting the ends of the arc to the center point), 
ArcType.CHORD  (close the shape by connecting the ends of the arc with a line segment) 
and ArcType.OPEN  (leave the shape open).

Angle 0° is at the right axis, 90° is at the top, 180° is at the left axis, and 270° is at the 
bottom. Positive values for length draw the arc in a counter-clockwise direction and 
negative values draw the arc in a clockwise direction.

Ellipse {
    centerX: 50, centerY: 50
    radiusX: 50, radiusY: 25
    fill: Color.CORAL
    stroke: Color.BLUE
}

Figure 4.9 Ellipse example

Figure 4.10 show three arc types and the code that creates type ArcType.ROUND . The arc 
is centered at point (50,50 ), with a y-axis radius of 25 and an x-axis radius of 50. The 
sweep of the arc begins at angle 45 and extends to angle 315 (a sweep of 270 degrees).

Polygon

Polygons have a sequence of x-coordinate and y-coordinate values for property 
points . A Polygon is a closed shape, so the last point connects to the first point with a 
line. Figure 4.11 shows an example object literal that renders a •stop signŽ octagon.

QuadCurve

The QuadCurve class defines a quadratic Bézier parametric curve segment with a 
starting and ending point ( startX , startY  and endX, endY) using a Bézier control point 
(controlX , controlY ). Figure 4.12 shows an example object literal and the resulting 

Arc {
    fill: Color.CORAL
    stroke: Color.BLUE
    centerX: 50 centerY: 50
    radiusX: 50 radiusY: 25
    startAngle: 45
    length: 270
    type: ArcType.ROUND
}

Figure 4.10  Arc examples

ArcType.ROUND

ArcType.CHORD

ArcType.OPEN

0°

90°

180°

270°

Polygon {
    points: [
        20.0,  0.0, 40.0,  0.0,
        60.0, 20.0, 60.0, 40.0,
        40.0, 60.0, 20.0, 60.0,
         0.0, 40.0, 0.0, 20.0,
    ]
    fill: Color.CORAL
    stroke: Color.BLUE
}

Figure 4.11 Polygon example

QuadCurve that the code renders. Point A is the starting point ( startX , startY ), point 
B is the control point (controlX , controlY ), and C is the ending point (endX, endY).

CubicCurve

The CubicCurve class defines a cubic Bézier parametric curve segment that intersects 
both a starting and ending point ( startX , startY  and endX, endY) using the specified 
Bézier control points (controlX1 , controlY1  and controlX2 , controlY2 ). Figure 4.13 
shows an example object literal and the resulting CubicCurve that the code renders. 
Point A is the starting point ( startX , startY ), points B and C are the control points 
(controlX1 , controlY1  and controlX2 , controlY2 ), and point D is the ending point 
(endX, endY).

Line

Lines have starting and ending x and y coordinate points. Property fill  defaults to 
null. Figure 4.14 shows an example object literal and the resulting Line that the code 
renders. Here, property strokeDashArray  renders a dashed line.

QuadCurve {
    startX: 0.0, startY: 100.0
    endX: 50.0, endY: 100.0
    controlX: 25.0, controlY: 0.0
    fill: Color.CORAL
    stroke: Color.BLUE
}

Figure 4.12 QuadCurve example

A

B

C

CubicCurve {
    startX: 0, startY: 50
    controlX1: 25, controlY1: -50
    controlX2: 75, controlY2: 150
    endX: 100, endY: 50
    fill: Color.CORAL
    stroke: Color.BLUE
}

Figure 4.13 CubicCurve example 

A

B

C

D

Polyline

Polylines have a sequence of x-coordinate and y-coordinate values for property 
points . Polyline is similar to Polygon, except that it does not automatically close. 
Figure 4.15 shows an example object literal that renders the Polyline shown. Note that 
the example does not specify a closing side and none is drawn.

SVGPath

SVGPath constructs a shape by parsing SVG (Scalable Vector Graphics) path data 
from a String.2 SVGPath parses only SVG path data, which includes commands and 
point data. 

2. SVG is a standard for specifying graphics. See http://www.w3.org/TR/SVG/paths.html  for 
a reference on SVG paths.

Line {
    startX: 10, startY: 100
    endX: 200, endY: 20
    stroke: Color.BLUE
    strokeDashArray: [16, 8]
}

Figure 4.14 Line example 

Polyline {
    points: [
        20.0,  0.0, 40.0,  0.0,
        60.0, 20.0, 60.0, 40.0,
        40.0, 60.0, 20.0, 60.0,
        0.0, 40.0, 0.0, 20.0,
    ]
    fill: Color.CORAL
    stroke: Color.BLUE
    strokeWidth: 6
}

Figure 4.15 Polyline example 

Polyline is not closed

Table 4.6 lists the available SVGPath commands. Capital letter commands indicate 
coordinate values with absolute position and lower case letter commands mean coor-
dinate values with relative position.  

Figure 4.16 shows an example object literal that renders the SVGPath shown. This 
draws a quadratic Bézier curve similar to the QuadCurve object in Figure 4.12 on 
page 110. 

Figure 4.17 shows an SVGPath that produces a spiral graphic. The code that renders 
this spiral is shown in Listing 4.4. 

TABLE 4.6 SVGPath Commands 

Command Description

Absolute Relative

M m move to

L l line to

H h horizontal line to

V v vertical line to

C c curve to

S s smooth curve to

Q q quadratic Bézier curve to

T t smooth quadratic Bézier curve to

A a elliptical arc

Z z close path

SVGPath {
    fill: Color.CORAL
    stroke: Color.BLUE
    strokeWidth: 2
    content : "M0 100"
    "Q25.0 0.0 50 100"
}

Figure 4.16 SVGPath example

A

B

C

Listing 4.4 SVGPath that produces a spiral

SVGPath {
    fill: Color.CORAL
    stroke: Color.BLUE
    strokeWidth: 2
    content : "M153 334"

"C153 334 151 334 151 334"
"C151 339 153 344 156 344"
"C164 344 171 339 171 334"
"C171 322 164 314 156 314"
"C142 314 131 322 131 334"
"C131 350 142 364 156 364"
"C175 364 191 350 191 334"
"C191 311 175 294 156 294"
"C131 294 111 311 111 334"
"C111 361 131 384 156 384"
"C186 384 211 361 211 334"
"C211 300 186 274 156 274"

}

ShapeIntersect/ShapeSubtract

ShapeIntersect and ShapeSubtract are composite shapes. You create them by combin-
ing two shapes specified in properties a and b (building block shapes). Building block 
shapes do not accept Shape properties such as fill  or stroke . However, composite 
shapes do„they are like any other Shape. Figure 4.18 shows several views illustrating 
ShapeIntersect and ShapeSubtract. 

View A shows the two building block shapes: Rectangle for property a and Ellipse for 
property b.

View B shows ShapeIntersect. The composite shape includes all points in the coordi-
nate space that are in both shapes.

Figure 4.17 SVGPath that creates a spiral

View C shows ShapeSubtract. Here, the composite shape consists of shape a minus 
shapeb.

Listing 4.5 shows the code that renders the views in Figure 4.18.

Listing 4.5 ShapeIntersect and ShapeSubtract

// View A
def g1 = Group {
    content: [
        Rectangle {
            width: 100
            height: 50

 stroke: Color.BLUE
            fill: Color.LIGHTSLATEGRAY
        }
        Ellipse {
            centerX: 100
            centerY: 25
            radiusX: 50
            radiusY: 25

stroke: Color.DARKBLUE
            fill: Color.web("#333333", .5)
        }
    ]
}
// View B
def shapeintersect = ShapeIntersect {
    stroke: Color.DARKCYAN
    a: Rectangle {
        width: 100
        height: 50
    }
    b: Ellipse {

Figure 4.18 ShapeIntersect and ShapeSubtract

View A: Rectangle and Ellipse source shapes

View B: ShapeIntersect (only common points)

View C: ShapeSubtract (Rectangle minus Ellipse)

        centerX: 100
        centerY: 25
        radiusX: 50
        radiusY: 25
    }
}
// View C
def shapeminus = ShapeSubtract {
    stroke: Color.DARKCYAN
    a: Rectangle { 
        width: 100
        height: 50
    }
    b: Ellipse { 
        centerX: 100
        centerY: 25
        radiusX: 50
        radiusY: 25
    }
}
. . .
VBox {

content: [ g1, shapeintersect, shapeminus ]
}
. . .

Text

Text is a specialized shape for displaying text. Table 4.7 lists the properties for compo-
nent Text.

TABLE 4.7 Text Properties 

Property Type Description

content String Text to be displayed.

font Font Font used to display text.

overline Boolean If true, each text line has a line above it.

strikethrough Boolean If true, each text line has a line through it.

underline Boolean If true, each text line has a line under it.

wrappingWidth Number Width constraint. If > 0, text is wrapped at word 
boundaries to comply.

textAlignment TextAlignment Horizontal text alignment; one of TextAlign-
ment.CENTER, TextAlignment.JUSTIFY , Text-
Alignment.LEFT  (default), and 
TextAlignment.RIGHT

TextOrigin.BASELINE  aligns the baseline of the text with the y coordinate value, which 
does not include the space taken by descending characters (such as lower case •gŽ or 
•yŽ). TextOrigin.BOTTOM  aligns the bottom of the text with the y coordinate value and 
includes space taken by descending characters. TextOrigin.TOP  aligns the top of the 
text with the y coordinate value. 

Since Text is a Node, you can easily implement a Text component that responds to an 
event. For example, Figure 4.19 shows a Text component used as a window close but-
ton. This code is useful in a run time environment that does not include a browser.

4.3  Paths

A Path lets you string together a sequence of path elements to define arbitrary shapes. 
Use Path when you need to customize a shape beyond the standard JavaFX shapes. 

textOrigin TextOrigin Aligns text with the y coordinate value in local 
coordinates; one of TextOrigin.BASELINE
(default), TextOrigin.BOTTOM , TextOri-
gin.TOP

x Number X coordinate of text origin.

y Number Y coordinate of text origin.

TABLE 4.7 Text Properties (Continued)

Property Type Description

def closeButton = Text {
    content: "X"
    fill: Color.BLACK
    x: 300
    y: 10
    textOrigin: TextOrigin.TOP
    font: Font {
        name: "Bitstream Vera Sans Bold"
        size: 18
    }
    cursor: Cursor.HAND
    visible: bind ("{__PROFILE__}" != "browser")
    onMousePressed: function(e: MouseEvent): Void {
        FX.exit();
    }
}

Figure 4.19 Text component as a close button



Paths 117

You can also use Path to create a PathTransition. This is a specialized animation that 
lets you move an object along a predefined path.

A Path has path elements. The first path element is always MoveTo. By default, coor-
dinates are expressed in absolute coordinate values. Set property absolute  to false for 
relative coordinate values. You•ll note that many JavaFX shapes have corresponding 
path elements (Arc is the shape and ArcTo is the path element). 

Table 4.8 lists the standard path elements.

Figure 4.20 shows a Path that forms an elliptical shape and Listing 4.6 is the code that 
renders it. We first create a sequence of path elements (pathElements ) and then use 
them to build a Path. The MoveTo element starts the path at point A, drawing an arc 
in the counter clockwise direction ( sweepFlag  is false ) to point B. A second arc starts 
at point B and finishes back at point A.

TABLE 4.8 Path Elements 

Path Element Description

MoveTo Move to specified x and y point. Required first path element.

ArcTo Forms an arc from the current coordinates to the specified x and y
coordinates using radius . Boolean property sweepFlag  specifies 
clockwise (if true ) or counter clockwise (if false ) sweep direction.

HLineTo Creates a horizontal line from the current point to x.

VLineTo Creates a vertical line from the current point to y.

LineTo Creates a line by drawing a line from the current coordinate to the 
new coordinates.

QuadCurveTo Creates curved path element, defined by two new points, by draw-
ing a Quadratic Bézier curve that intersects both the current coordi-
nates and the specified coordinates (x, y), using the specified point 
(controlX , controlY )  as a Bézier control point.

CubicCurveTo Create curved path element, defined by three new points, by draw-
ing a Cubic Bézier curve that intersects both the current coordinates 
and the specified coordinates (x, y), using the specified points 
(controlX1 , controlY1 ) and (controlX2 , controlY2 ) as Bézier 
control points.

ClosePath Closes current path by drawing a line from the current point to the 
starting point.

Listing 4.6 Path and PathElements

def centerX = 100;
def centerY = 100;
def radiusX = 25;
def radiusY = 50;

// this provides the path elements for the Path
def pathElements = [
    MoveTo {
        x: centerX
        y: centerY + radiusY
    }
    ArcTo {
        x: centerX
        y: centerY - radiusY
        radiusX: radiusX
        radiusY: radiusY
    }
    ArcTo {
        x: centerX
        y: centerY + radiusY
        radiusX: radiusX
        radiusY: radiusY
    }
];

def path = Path {
    stroke: Color.DARKGRAY
    strokeWidth: 2
    elements: pathElements
}

JavaFX Tip

With Path you can define a PathTransition, an animation of a node along a sequence of path 
elements. See •Path AnimationŽ on page 233.

Figure 4.20 Path example using MoveTo and ArcTo
A

B



Layout Components 119

Book Examples

€ •Main PathElementsŽ, Listing 7.24 on page 242, (Chapter 7). Creates an animation 
path with CubicCurveTo, LineTo, and MoveTo elements.

€ •PathElements that define the PathTransitionŽ, Listing 8.17 on page 280, 
(Chapter 8). Creates a photo carousel path with MoveTo and ArcTo elements.

4.4  Layout Components

Layout components specialize Container and Group (see Figure 4.2 on page 97). They 
manage their content, which can be a combination of one or more Shapes, Groups, or 
other layout nodes that make it convenient to arrange nodes in a scene. Table 4.9 
describes the layout components available in JavaFX.

TABLE 4.9 Layout Components

Component Description

HBox Provides a horizontal layout of its contents in a single row. Con-
trol space between subnodes with property spacing . Property 
hPos  is the horizontal position of the row of nodes within this con-
tainer's width. Property vPos  is the vertical position of each node 
within the space allocated to it in the row.

VBox Provides a vertical layout of its contents in a single column. Con-
trol space between subnodes with property spacing . Property 
hPos  is the horizontal position of each node within the space allo-
cated to it in the column. Property vPos  is the vertical position of 
the column of nodes within this container's height.

Flow Provides a layout of its contents in either a horizontal or vertical 
flow (vertical flow if property vertical  is true), wrapping at its 
current width or height boundaries (properties width , height ).
Use hGap, vGap for spacing. Properties hPos , vPos  depend on ver-
tical or horizontal flow. Each cell is sized to fit its content node.

Tile Provides a layout of its contents in either a horizontal or vertical 
flow (vertical flow if property vertical  is true), wrapping at its 
current width or height boundaries (properties width , height ).
Use hGap, vGap for spacing and hPos , vPos  to position contents 
within each cell. All cells are uniform size. Properties tileHeight
and tileWidth  control cell size.

Stack Arranges its content nodes in a back-to-front stack. Useful for tab-
style nodes or stacks of nodes, such as ImageView components.

HBox and VBox Layout Components

HBox and VBox are layout components that manage a sequence of nodes in property 
content . HBox provides a horizontal layout of its contents, one after the other, equally 
spaced and in a single line. (Content that extends beyond the boundaries of the scene 
is clipped.) Similarly, VBox provides a vertical layout of its contents. You can nest 
HBox and VBox components as needed, and you can control the amount of space 
between subnodes with property spacing . Use properties vPos  and hPos  to control 
subnode positioning and row/column position.

Figure 4.21 shows a VBox layout component that contains two HBox nodes (each with 
three rectangles) and a line between them. The rectangles in each HBox have different 
spacing. The VBox component is positioned with layoutX  and layoutY  properties.

Listing 4.7 shows the code that renders the scene graph from Figure 4.21.

Listing 4.7 HBox, VBox Layout Example

def colors = [ Color.RED, Color.BLUE, Color.GREEN ];

def rectangles = for (i in [0..5])
    Rectangle {
        width: 20
        height: 50
        fill: colors[i mod 3]
    }

. . .

VBox {
layoutX: 50
layoutY: 20

Figure 4.21 Using HBox and VBox layout components

spacing: 20
content: [

HBox {
spacing: 10
content: [ rectangles[0], rectangles[1], rectangles[2] ]

}
Line {

startX: 0
startY: 0
endX: 100
endY: 0
stroke: Color.BLACK
strokeWidth: 2

}
HBox {

spacing: 20
content: [ rectangles[3], rectangles[4], rectangles[5] ]

}
]

}
. . .

Book Examples

€ •Piano„Step 3: Add the keyboard„Part 2Ž, Listing 6.9 on page 183, (Chapter 6). 
Uses VBox to layout components.

€ •Piano„Step 4: Add Swing buttons and note buffer„Part 3Ž, Listing 6.12 on 
page 187, (Chapter 6). Uses HBox to layout components.

€ •Chutes and Ladders Scene GraphŽ, Listing 7.27 on page 246, (Chapter 7). Uses 
VBox and nested HBox to layout components.

Flow and Tile Layout Components

The Flow layout component provides a layout of its content in either a horizontal 
(default) or vertical flow (property vertical  is true ). Flow layout wraps content at its 
boundaries (specified by properties width  and height ). Properties hGap and vGap

define spacing, depending on whether the flow is horizontal or vertical. For example, 
property hGap defines the amount of horizontal space between each node in a horizon-
tal flow or the space between columns in a vertical flow. Likewise, properties hPos  and 
vPos  depend on vertical or horizontal flow. For horizontal flow, property hPos  defines 
the horizontal position of each flow within the Flow container's width. For vertical 
flow, property hPos  defines the horizontal position of nodes within each cell. Because 
each node follows the previous node, Flow layout does not necessarily produce a grid 
layout (see Figure 4.22).

The Tile layout component provides a layout of its content in uniformly sized spaces 
or •tiles.Ž Tile layout lets you specify the size of tiles with properties tileWidth  and 
tileHeight  and the number of columns and rows (properties columns  and rows ). Pro-
perties hGap and vGap define the amount of horizontal space between tiles in a row and 
vertical space between tiles in a column, respectively. Properties hPos and vPos  posi-
tion a node when the node doesn•t fill its space in a tile. (The default value for vPos  is 
VPos.CENTER and for hPos is HPos.CENTER.)

Figure 4.22 shows Flow layout (View A), Tile layout (View B), and Tile layout where 
vPos  is set to VPos.TOP (View C). Listing 4.8 lists the code for each view.

Listing 4.8 Flow and Tile Layout Examples

// View A: Flow Layout
Flow {

vertical: true
width: 200, height: 200
layoutX: 20, layoutY: 80
hGap: 20, vGap: 20
content: [ rectangles, circles ]

}
// View B: Tile Layout
Tile {

vertical: true
width: 200, height: 200
layoutX: 20, layoutY: 80
hGap: 20, vGap: 20
content: [ rectangles, circles ]

}
// View C: Tile Layout with vPos
Tile {

Figure 4.22 Flow and Tile layout components

View A: Flow Layout View B: Tile Layout View C: Tile Layout
vPos: VPos.TOP 

vPos: VPos.TOP
vertical: true
width: 200, height: 200
layoutX: 20, layoutY: 80
hGap: 20, vGap: 20
content: [ rectangles, circles ]

}

Book Example

€ •Piano„Step 4: Add Swing buttons and note buffer„Part 3Ž, Listing 6.12 on 
page 187, (Chapter 6). Uses Flow to layout components.

Stack Layout Component

The Stack layout component arranges its content nodes in a back-to-front stack. That 
is, each node is •stackedŽ on top of previously added nodes. Stack layouts are useful 
for tab-style nodes or stacks of nodes, such as ImageView panels (perhaps a slide 
show). Figure 4.23 shows an example with a tab-style panel selection mechanism. 
Selecting the tab brings that panel to the top of the stack.

Listing 4.9 shows the code that renders the layout in Figure 4.23. The tab has an 
onMouseClicked  event handler to move its node to the front with function 
group.toFront .

Figure 4.23 Stack layout component

Listing 4.9 Stack Layout Example

// Build three panels consisting of a tab, tab body, text label, and Text
def panels = for (i in [0..2]) {
    var tab: Rectangle;
    var tabBody: Rectangle;
    var tabLabel: Text;
    var group: Group;
    group = Group {
        effect: DropShadow { }
        content: [
            tab = Rectangle {

. . .
   cursor: Cursor.HAND
   stroke: Color.BLACK

onMouseClicked: function(e: MouseEvent): Void {
      group.toFront();

                }
     arcWidth: 15

    arcHeight: 15
            }
            tabBody = Rectangle { . . . }
            tabLabel = Text { . . . }
            Text { . . . }
        ]
    }
}

// put panels in a Stack layout component
Stack {

// reverse order so that panels[0] is on top
content: [ panels[2], panels[1], panels[0]  ]

}

4.5  Geometry

A scene is a two-dimensional coordinate space with nodes. JavaFX provides several 
convenience classes to help you manage this geometry. You also have •boundingŽ 
rectangles to manage the relative size and placement of nodes.

Point2D

Class Point2D is a convenience class that encapsulates the x and y coordinates for 
points. You initialize a point with public-init  properties x and y (both Numbers). The 
Point2D member function distance  calculates the distance between two points. Here 
are some examples.

var origin = Point2D { x: 0, y: 0 }

var newPoint = Point2D { x: 3, y: 4 }
println("distance from origin = {newPoint.distance(origin)}"); // 5.0
println("distance from origin = {newPoint.distance(0,0)}"); // 5.0

Bounds/Rectangle2D

Class Bounds (Rectangle2D prior to JavaFX 1.2) is a convenience class that encapsu-
lates geometry properties for bounding rectangles.3 Bounds lets you query the posi-
tion or bounds of objects in your scene graph. It•s particularly useful when centering 
or positioning objects based on the dimensions or positions of other objects. 

Table 4.10 lists the Bounds properties. Each graphical object has multiple bounding 
rectangles that define its size and position. 

Bounding Rectangles

When you add an object (Node) to a scene graph, JavaFX provides four bounding 
rectangles that describe the node•s position and size. Two (boundsInLocal  and layout-

Bounds ) are relative to the local coordinate system (the current container). Bounding 
rectangle boundsInParent  is relative to the coordinate system of the parent node. 
Bounding rectangle boundsInScene  is relative to the coordinate system of the scene. 
Table 4.11 lists the properties for these node bounding rectangles, which are all 
Bounds objects.

Typically, you•ll want to use bounding rectangles for position and size information 
instead of a node•s own dimensions or position properties. Which bounding rectangle 
you use depends on the bounding rectangle•s node and your application. Neither 
boundsInLocal  or layoutBounds  responds to the effects of layoutX , layoutY , translateX ,
translateY , rotate , scaleX  and scaleY .

3. Beginning with JavaFX 1.2, class Bounds is used to describe the bounds of a node or other 
scene graph object. (Rectangle2D is still used to define the viewport of ImageView and 
MediaView.)

TABLE 4.10 Bounds/Rectangle2D Properties 

Property Type Description

height Number Height of bounding rectangle

width Number Width of bounding rectangle

minX Number Left x value of bounding rectangle

minY Number Top y value of bounding rectangle

maxX Number Right x value of bounding rectangle

maxY Number Bottom y value of bounding rectangle

Bounding rectangle boundsInParent  is relative to the parent node and responds to all 
transformations on the node, including those set in transforms , layoutX , layoutY ,
translateX , translateY , scaleX , scaleY , and rotate  properties. 

Bounding rectangle boundsInScene  provides the same dimensions as boundsInParent .
However, the bounding rectangle is defined in terms of the scene•s coordinate system. 
This means coordinate values will change if the parent node has been adjusted with 
properties such as layoutX  and layoutY .

Layout Tip

When doing layout, you almost always want to use layoutBounds  instead of boundsInScene

andboundsInParent  (layoutBounds  is more efficient). Also, it is an error to bind properties 
such as centerX  and centerY  (for Circles) or x and y (for Rectangle or Text) to an expression 
that depends upon its own layoutBounds for the purpose of positioning the node.

Positioning with Bounding Rectangles

Suppose you want to position a VBox layout component over a background Rectangle 
node. Figure 4.24 shows a vertical box (VBox) layout component containing three Text 
components (•Small,Ž •Medium,Ž and •LargeŽ). The VBox is positioned (using lay-

outX  and layoutY ) so that it is on top of the background rectangle.

TABLE 4.11 Node Bounding Rectangles 

Property Type Description

boundsInLocal Bounds

(Rectangle2D
prior to JavaFX 
1.2)

Rectangular bounds of Node in local coordinate 
space, including space for stroke, effect, clip. 
Not included is rotate, transforms, layout, 
translate, scale properties. 

layoutBounds Bounds

(Rectangle2D
prior to JavaFX 
1.2)

Rectangular bounds of Node. Includes trans-
forms, but not rotate, translate, layout, scale 
properties.

boundsInParent Bounds

(Rectangle2D
prior to JavaFX 
1.2)

Rectangular bounds of Node in parent coordi-
nate system; includes effects of transforms, lay-
out, translate, scale, and rotate.

boundsInScene Bounds

(Rectangle2D
prior to JavaFX 
1.2)

Rectangular bounds of Node in scene coordi-
nate system; includes effects of transforms, lay-
out, translate, scale, and rotate.

Listing 4.10 shows the code that renders the scene in Figure 4.24. 

To align the VBox directly over the Rectangle, set the VBox properties layoutX  and 
layoutY  with the background rectangle•s layoutBounds.minX  and layoutBounds.minY ,
respectively. Read-only variable margin  maintains spacing so that the three subnode 
Text components aren•t directly at the top and left edges. 

You must set Text property textOrigin  to TextOrigin.TOP . Otherwise, the default 
placement of Text aligns the Text baseline with the top of VBox. With TextOrigin.TOP ,
the top of the Text aligns with the top of the VBox.

Note that you position the background Rectangle with properties x and y. The VBox 
component is positioned using layoutX  and layoutY .

Listing 4.10 Using Bounding Rectangles for Layout

def controlWidth = 150; 
def controlHeight = 150;
def offsetX = 20; 
def offsetY = 20;

def background = Rectangle {
    width: controlWidth
    height: controlHeight
    x: offsetX y: offsetY
    arcWidth: 20 arcHeight: 20
    stroke: Color.SLATEBLUE
    strokeWidth: 2
    fill: Color.ANTIQUEWHITE
}

def margin = 10;
def vbox = VBox {
    spacing: 20

// Use layoutX and layoutY (new to JavaFX 1.2) to position nodes

Figure 4.24 Using bounding rectangles to position two nodes

background RectangleVBox

    layoutX: background.layoutBounds.minX + margin
    layoutY: background.layoutBounds.minX + margin
    content: [
        Text {
            content: "Small"
            textOrigin: TextOrigin.TOP
        }
        Text {
            content: "Medium"
            textOrigin: TextOrigin.TOP
        }
        Text {
            content: "Large"
            textOrigin: TextOrigin.TOP
        }
    ]
}

Book Example

€ •CarouselPhoto Scene GraphŽ, Listing 8.15 on page 278, (Chapter 8). Uses bound-
ing rectangles to position images and image titles.

5 User Interface 
Components

Graphical objects include user interface components, described in this chapter. JavaFX 
currently offers two flavors: Java Swing-based components that are wrapped for 
seamless integration into JavaFX and •nativeŽ JavaFX components. Recently, the 
native JavaFX UI component offerings have increased, and we briefly describe these 
new components in this chapter. We point you to examples throughout the book that 
use the described components, where possible.

You can apply CSS styles to graphical objects (•skinŽ them), which is especially useful 
for incorporating UI components with a uniform look and feel. You can also build 
your own •skinnableŽ UI components. (Skinnable components let you apply a uni-
form style to components through external CSS style sheets. By using different style 
sheets, you can change the entire look of your components.)

What You Will Learn

€ JavaFX user interface components

€ Swing-based user interface components

€ Integrating UI controls into applications with binding

€ Applying CSS styles to graphical objects

€ Creating •skinnableŽ UI components

5.1  JavaFX UI Controls

JavaFX includes a native TextBox UI component which is discussed below. The JavaFX 
1.2 release also includes additional UI components briefly described in Table 5.2.

TextBox

A TextBox obtains textual input from users. You specify the width with property col-

umns and an event handler with property action . Boolean editable  specifies whether 
or not the TextBox is editable by the user (true  is the default). Property rawText

reflects the text in the TextBox (updated as the user provides input) and text  holds the 
input when the event handler is invoked. In addition, properties adjustingSelection ,
dot , and mark  let you query user text selection. Figure 5.1 shows an example. 

Table 5.1 lists the TextBox properties.

TABLE 5.1  TextBox Properties 

Property Type Description (Default)

action function():Void Function that is called when an action is 
fired on this TextBox (typically with 
<Enter>).

columns Number Horizontal size: approximate number of 
characters (default 10). 

editable Boolean If true, text is editable by user (default true).

font Font Default font for text.

selectOnFocus Boolean If true, focus gain on this TextBox via key-
board or softkey navigation causes a selec-
tion of its text contents (default true).

text String Text contained in this TextBox, updated 
when user commits.

rawText String Raw text value (as user types).

promptText String Displayed when no text has been entered 
into the Control.

adjustingSelection

dot

mark

Boolean

Integer

Integer

Used to query state of text selection. Prop-
erty dot  is location of caret, mark  is the 
anchor. If dot != mark  then there is text 
selection.

var textInput: TextBox;
. . .

textInput = TextBox {
columns: 20
action: function(): Void {

if (textInput.text != "") {
processInput(textInput.text);

}
}

}

Figure 5.1 TextBox UI Component

Book Examples

€ •FlickrTag (Main.fx)Ž, Listing 9.20 on page 310, (Chapter 9). Provide a search tag 
for Flickr.

€ •Scene Graph (FlickrUser)Ž, Listing 9.24 on page 314, (Chapter 9). Provide a screen 
name for Flickr.

UI Components

Table 5.2 lists the UI components included with the JavaFX 1.2 release. With these 
components, developers will have a richer selection of built-in UI controls. Unlike the 
JavaFX Swing components (which only run in the desktop environment), these UI 
components are portable across all environments (desktop, mobile, and TV). See 
Figure 1.1, •JavaFX Platform,Ž on page 2 for an overview of the JavaFX platform. 

TABLE 5.2  UI Components 

UI Component Description

TextBox Textual input with action event handler.

Button Pressable component with action event handler.

Hyperlink Alternative view of Button. Keeps track of whether it has been 
•visitedŽ and optionally renders differently if it has.

ToggleButton Similar to SwingToggleButton. If the component is part of a Tog-
gleGroup, the group ensures that only one ToggleButton is 
selected at a time. Otherwise, toggles between selected and not 
selected states.

RadioButton Extends ToggleButton and provides alternate view (the tradi-
tional radio button look).

ToggleGroup Groups ToggleButtons or RadioButtons so that selection is mutu-
ally exclusive among all buttons in the same group.

CheckBox Tri-state selection control typically skinned as a box with a check-
mark or tick mark when checked.

ListView Scrollable list of selectable items.

ComboBox List of pre-defined Strings with the option of entering a custom 
string.

Label Non-editable text and/or graphic control. Displayed text is modi-
fied to fit within a specific space using ellipses or truncation. 

ScrollBar Scrolling control.

Slider Similar to SwingSlider.

Figure 5.2 shows many of these UI components and Listing 5.1 is the source code for 
the application shown.

Listing 5.1 UI Component Menagerie

// Two ToggleButtons in a ToggleGroup and One CheckBox
var toggleGroup = ToggleGroup {}
def tg1 = ToggleButton { text: "First", toggleGroup: toggleGroup }
def tg2 = ToggleButton { text: "Second", toggleGroup: toggleGroup }
def tg3 = CheckBox { text: "CheckBox One" }

ProgressBar Visual progress status indicator. Property progress  (if between 0 
and 1) indicates percentage complete, if >= 1 indicates done.

ProgressIndicator Alternate view to ProgressBar. Provides a small, circular progress 
indicator.

TABLE 5.2  UI Components (Continued)

UI Component Description

Figure 5.2 UI component menagerie

ToggleButtons

RadioButtons

Slider

Buttons

Label with text

ProgressBar

ProgressIndicator

(attached to Rectangle opacity)

ListView

Hyperlinks

CheckBox

and graphic

// Three RadioButtons in a ToggleGroup
def tb = ToggleGroup {}
def r1 = RadioButton { text: "Small", toggleGroup: tb }
def r2 = RadioButton { text: "Medium", toggleGroup: tb }
def r3 = RadioButton { text: "Large", toggleGroup: tb }

// Slider controls Rectangle•s opacity
def slider = Slider {
    min: 0, max: 1
    vertical: false
    showTickMarks: true, showTickLabels: true
    value: 0.5
    majorTickUnit: .25, minorTickCount: 3
}
def rectangle = Rectangle {
    width: 50, height: 80
    opacity: bind slider.value, fill: Color.BLUE
}

// Two Buttons
def b1 = Button {
    text: "Click Button 1"
    strong: true
    action: function() { println("Button 1 clicked"); }
}
def b2 = Button {
    strong: false
    text: "Click Me Button 2"
    action: function() { println("Button 2 clicked"); }
}

// Two Hyperlinks and a Label
def h1 = Hyperlink { text: "Hyperlink 1" }
def h2: Hyperlink = Hyperlink { text: "Hyperlink 2" }
def l1 = Label { text: "Hi, I am a label." font: Font { ... }
    graphic: ImageView { image: Image { url: "{__DIR__}duke.gif" } }
    textFill: Color.CRIMSON
}

// ProgressBar and ProgressIndicator reporting on same variable n
var n: Number = 0;
var pb = ProgressBar {
    progress: bind ProgressBar.computeProgress(1000, n)
}
var pi = ProgressIndicator {
    progress: bind ProgressIndicator.computeProgress(1000, n)
}
def lv = ListView { items: ["One", "Two", "Three"] }

Popup Windows

JavaFX includes two external •windowsŽ you can invoke from applications: Alert and 
Popup. Both components are not part of your main application•s scene graph but have 
their own stage.

Alert provides a high-level, configurable popup dialog. There are three types: con-
firm, inform, and question, instantiated by invoking either Alert function confirm ,
inform , or question . The title is the optional first argument. Figure 5.3 shows a ques-
tion alert with Yes/No buttons.

Popup provides a general-purpose component that pops up on the user•s screen. You 
can build a Popup with any scene graph structure that you•d like. Figure 5.4 shows a 
Popup rendered with a Rectangle and Text component.

Figure 5.3 Alert window (question)

// action property in Button
action: function() {

if (Alert.question (
"Important Question???",
"Do you want to continue?"))

statusString = "We will continue!"
else

statusString = "We will stop!";
}

}

Figure 5.4 Popup window

def myPopUp = Popup {
// put Popup contents here

    content: [ . . . ]
}
. . .
// action property in Button
action: function() {

if (myPopUp.visible) myPopUp.hide()
// use absolute screen coordinates
else myPopUp.show(600, 400);

}

5.2  Swing Components

JavaFX provides direct access to a subset of the Java Swing components. These Swing 
components have been wrapped so they are consistent with the JavaFX scene graph 
model and node behavior. Swing components only run in the desktop environment, 
which means you can•t use them in mobile applications.

SwingButton

SwingButton provides a push button graphical UI component. You can adorn it with 
text, an icon, or both. Use property action  for event handling, as shown in Figure 5.5. 

Book Examples

€ •Piano„Step 4: Add Swing buttons and note buffer„Part 3Ž, Listing 6.12 on 
page 187, (Chapter 6). Uses SwingButton components to control a piano keyboard.

€ •Photo4: Photo Study Scene GraphŽ, Listing 8.7 on page 261, (Chapter 8). Uses 
SwingButton components to select effects to apply to a photo.

SwingCheckBox

SwingCheckBox is a •checkboxŽ component that lets users toggle between checked 
and unchecked to represent boolean values. Property selected  is true when the box is 
checked. You can enforce the behavior of checking only one SwingCheckBox compo-
nent in a group by assigning the same value for property toggleGroup  to all Swing-
CheckBox components in the group. (Figure 5.11 on page 140 shows how to use 
property toggleGroup  and enforce mutual exclusion behavior.)

Figure 5.6 illustrates a SwingCheckBox component with an action event handler. A 
node (p) moves to the front of its container if the checkbox is selected. Otherwise, it 
moves to the back. 

SwingButton {
    text: "Group Rotate"
    action: function(): Void {
        shapeArea.rotate += 45;
    }
}

Figure 5.5 SwingButton example

SwingComboBox

SwingComboBox provides a selection mechanism from a drop-down list. When you 
make the combo box editable, users can type values into an editable field. You popu-
late the SwingComboBox with a sequence of SwingComboBoxItem components in 
property items . Figure 5.7 shows a SwingComboBox populated with four SwingCom-
boBoxItem selection options.

SwingComboBoxItem

The items in a SwingComboBox are of type SwingComboBoxItem. Property text  pro-
vides a text-based name for the item. You can construct an arbitrary object with prop-
erty value . For example, Listing 5.2 builds a different sized Font object for each 
SwingComboBoxItem providing a convenient way to update an object•s font  property 
with binding. The Text object•s font  property bind expression is shown in bold.

def frontCheck: SwingCheckBox = SwingCheckBox {
    text: "Polygon In Front"
    selected: true
    action: function(): Void {
        if (frontCheck.selected) { p.toFront(); }
        else { p.toBack(); }
    }
}

Figure 5.6 SwingCheckBox example

Figure 5.7 SwingComboBox and SwingComboBoxItems

Listing 5.2 SwingComboBox and SwingComboBoxItem

def fontbox = SwingComboBox {
    width: 100
    items: [
        SwingComboBoxItem {
            text: "12"
            value: Font { size: 12 }
            selected: true
        }
        SwingComboBoxItem {
            text: "14"
            value: Font { size: 14 }
        }
        SwingComboBoxItem {
            text: "18"
            value: Font { size: 18 }
        } ]
}
. . .
Text {
    x: 35, y: 30
    font: bind fontbox.selectedItem.value as Font
    content: "The quick brown fox jumps over the lazy yellow dog."
}

SwingIcon

SwingIcon is an image for decorating components. Use JavaFX Image to set the icon•s 
image. Figure 5.8 shows how to create a button that includes both an icon and text. 

SwingLabel

SwingLabel components display a string, icon, or both. Use property icon  (see 
Figure 5.8) to supply images. Figure 5.9 shows a text-only SwingLabel component 
associated with SwingSlider component fillOpacity  (see •SwingSliderŽ on page 141). 

SwingButton {
text: "Duke says 'Hi!'"
icon: SwingIcon { 

image: Image { url: "{__DIR__}duke.gif" } }
action: function(): Void {

// action code here
}

}

Figure 5.8 SwingButton with SwingIcon Example

Property labelFor  lets you optionally associate a label with a swing component. 
(Property labelFor  is null if this label is not associated with a component.)

SwingScrollPane

SwingScrollPane is a container that allows its contents to be scrolled. Set property 
view  to its scrollable contents. Figure 5.10 shows a SwingScrollPane component (with 
a SwingList) and Listing 5.3 shows the corresponding code that renders it.

SwingList

SwingList displays a list of SwingListItem objects and allows users to select one or 
more items. Property items  holds a sequence of SwingListItem objects. Property 
selectedIndex  returns the index of the selected item, and property selectedItem

returns the selected item. When multiple items are selected, selectedIndex  is the 
smallest selected index and returns -1 if there is no selection.

SwingListItem

SwingListItem represents a selectable item in a SwingList component. Property 
selected is true  when this item has been selected. Property text  contains the display 
text and property value  is an object associated with this list item. 

SwingLabel {
width: controlWidth - 20
text: " Polygon Opacity"
labelFor: fillOpacity

}
fillOpacity

Figure 5.9 SwingLabel Example

Figure 5.10 SwingScrollPane, SwingList, and SwingListItems

Listing 5.3 shows an example with a SwingListItem. Here, property value  is an Effect 
object (DropShadow, InnerShadow, Glow, MotionBlur) that updates a Rectangle using 
binding. The Rectangle•s effect  property binding expression is in bold.

Listing 5.3 SwingScrollPane, SwingList, and SwingListItems

var effectList: SwingList;
SwingScrollPane {
    height: 100
    width: 150
    scrollable: true
    view: effectList = SwingList { items: [

  SwingListItem {
    text: "No Effect"
    selected: true

     value: null
            },

  SwingListItem {
   text: "Drop Shadow"

  value: DropShadow {
                }
            }

  SwingListItem {
   text: "Inner Shadow"

  value: InnerShadow {
       offsetX: 4
       offsetY: 4

                }
            }

  SwingListItem {
     text: "Glow"

    value: Glow {
       level: 1

                }
            },

  SwingListItem {
   text: "Motion Blur"

  value: MotionBlur {
       angle: 45
       radius: 10

                }
            }

. . .
        ]
    }
}
. . . 
// Rectangle that receives effect from SwingList component selection
Rectangle {
    x: 30
    y: 10

    height: 100
    width: 80
    fill: defaultColor
    stroke: Color.BLACK

effect: bind effectList.selectedItem.value as Effect
}

SwingRadioButton

SwingRadioButton renders a selectable radio button. If you have multiple SwingRa-
dioButton components that share the same SwingToggleGroup (as set in property 
toggleGroup ), the SwingToggleGroup enforces mutual exclusion selection. Figure 5.11 
shows a radio button group that exhibits this behavior.

SwingToggleButton

SwingToggleButton provides the same behavior as SwingRadioButton except that it 
looks like a regular button. A SwingToggleButton can be toggled between a pressed 
and released state to represent a boolean value to the user. If you assign multiple 

SwingRadioButton {
            toggleGroup: pizzaSizeGroup
            text: "Small"
            action: function(): Void {

// action . . .
            }
        }
SwingRadioButton {
            toggleGroup: pizzaSizeGroup
            text: "Medium"
            action: function(): Void {

// action . . .
            }
        }
SwingRadioButton {
            toggleGroup: pizzaSizeGroup
            selected: true
            text: "Large"
            action: function(): Void {

// action . . .
            }
        }

Figure 5.11 SwingRadioButton example with mutual exclusion

SwingToggleButton components to the same SwingToggleGroup, the group enforces 
mutual exclusion selection.

SwingToggleGroup

SwingToggleGroup is a class used to enforce mutual exclusion selection for a set of 
buttons (either SwingToggleButton, SwingCheckBox, or SwingRadioButton). When 
buttons are in the same SwingToggleGroup, selecting one button deselects all the 
other buttons with the same toggleGroup  setting.

SwingSlider

SwingSlider lets users graphically select a value by sliding a knob within a bounded 
interval. You provide properties maximum and minimum  for the upper and lower bounds 
of the slider. Property value  is the slider•s current value, which is within the maxi-
mum and minimum range (inclusive). If property vertical  is true, the slider is ren-
dered vertically. 

Figure 5.12 shows an example of SwingSlider that lets users set a node•s opacity. The 
slider•s value is in the range 0 to 100, but the binding expression divides by 100.0 to 
get the desired opacity range (0 to 1). Note that properties minimum , maximum, and value

are all Integers.

SwingTextField

SwingTextField allows the display and editing of a single line of text. If property edit-

able  is true, users can edit the text. Figure 5.13 shows an editable SwingTextField. 
Here, a Text component•s content  property is bound to the SwingTextField•s text

property. As the user types in text, the Text component reflects the new input.

def fillOpacity = SwingSlider {
    minimum: 0
    maximum: 100
    width: controlWidth - 20
    height: 20
    value: 50
}
. . .

var opacity = bind (fillOpacity.value / 100.0);

Figure 5.12 SwingSlider example with binding opacity

5.3  Swing Example

Figure 5.14 shows two views from an application called Order Your Pizza. Unfortu-
nately, this application doesn•t really order pizza for you, but it does illustrate several 
Swing components. The application uses radio buttons with mutual exclusion selec-
tion and checkboxes that allow multiple selections. You•ll also see how JavaFX bind-
ings simplify the integration of UI components into your applications.

The example shows how easy it is to build graphical applications above and beyond 
one that simply uses radio buttons, checkboxes, and text components. Here, for exam-
ple, you see visual pizzas and pizza toppings. As users select different sized pizzas, 
the pizza graphic changes size too. Furthermore, if the pizza includes one or more 
toppings, these toppings remain within the boundaries of the pizza.

Let•s begin with the graphical objects first: pizza and pizza toppings.

Pizzas Are Circles

Listing 5.4 shows the code that builds the pizza. Our pizza comes in three sizes (small, 
medium, and large with radius 40, 55, and 70, respectively). Pizzas have a crust 
(strokeWidth  is 8) and a drop shadow. Property radius  is bound to the sequence 
pizzaSize  via the variable pizzaSizeIndex .

Listing 5.4 also defines clipToppings , a circle that matches the three pizza sizes. The 
topping group (pepperoni, sausage, onions) uses this circle for its clip  property so 

def tf = SwingTextField {
    columns: 10
    editable: true
    text: "Now is the time"
}
. . .

Text {
    x: 35
    y: 30

content: bind tf.text
    fill: Color.BLACK
}

Figure 5.13 SwingTextField Example

that its size is adjusted appropriately when the user chooses a different sized pizza. 
Note that the node used for property clip  is not inserted into the scene graph; it sim-
ply provides a geometric reference for clipping behavior.

Listing 5.4 Pizza and Clip Circles

def pizzaCenterX = 80; 
def pizzaCenterY = 100;
def pizzaSize = [40, 55, 70];
var pizzaSizeIndex = 2; // start with a "Large"
def pizzaArea = Group { layoutX: 20 }

// This is the Pizza
insert Circle {
    centerX: pizzaCenterX
    centerY: pizzaCenterY
    radius: bind pizzaSize[pizzaSizeIndex]
    fill: Color.web("#D72D02")
    stroke: Color.web("#CB7621")
    strokeWidth: 8
    effect: DropShadow { }
} into pizzaArea.content;
// This circle is used to provide clipping reference for the toppings
def clipToppings = Circle {
    centerX: pizzaCenterX
    centerY: pizzaCenterY
    radius: bind pizzaSize[pizzaSizeIndex];
}

Figure 5.14 Order Your Pizza

Toppings Are Circles

Listing 5.5 displays the graphical objects that render pizza toppings. The toppings are 
in a group, with a clip  property bound to the currently selected pizza size. The top-
pings themselves are small circles built at the coordinate values in sequence toppin-

gLocations . Each pair of numbers initializes the centerX  and centerY  properties of a 
circle. The for  loop builds the toppings and inserts them into the topping  Group.

for (i in [0.. <sizeof toppingLocations step 6]) { . . . }

Each pizza topping has its own color and all have their visible  property bound to the 
SwingCheckBox component that selects the topping.

Listing 5.5 Toppings

// x and y coordinate values for toppings
def toppingLocations = [
    88, 42, 101, 45, 118, 58, 126, 72, 134, 88, 135, 108, 
    123, 128, 114, 141, 97, 150, 82, 152, 59, 151, 42, 146, 
    30, 131, 23, 118, 23, 96, 28, 75, 41, 58, 60, 45
    77, 53, 90, 60, 109, 69, 112, 84, 113, 100, 120, 115, 
    106, 130, 87, 141, 68, 137, 50, 123, 42, 109, 45, 89, 
    54, 72, 78, 65, 93, 70, 101, 81, 100, 96, 97, 108, 
    88, 124, 65, 118, 61, 108, 59, 95, 67, 85, 81, 85
    85, 98, 79, 105, 70, 105, 67, 94, 78, 94, 77, 100
];

// Group to hold the toppings, clipped to the current pizza size
def toppings = Group { clip: bind clipToppings }

// Build (small) circles for each kind of topping
for( i in [0.. <sizeof toppingLocations step 6]) {
    insert Circle {
        effect: DropShadow {}
        centerX: toppingLocations[i]
        centerY: toppingLocations[i + 1]
        radius: 5
        fill: Color.web("#D47F54");
        visible: bind checkPepperoni.selected
    } into toppings.content;
    insert Circle {
        effect: DropShadow {}
        centerX: toppingLocations[i + 2]
        centerY: toppingLocations[i + 3]
        radius: 5
        fill: Color.web("#FFAA00");
        visible: bind checkSausage.selected
    } into toppings.content;
    insert Circle {
        effect: DropShadow {}
        centerX: toppingLocations[i + 4]

        centerY: toppingLocations[i + 5]
        radius: 5
        fill: Color.BURLYWOOD;
        visible: bind checkOnions.selected
    } into toppings.content;
}
insert toppings into pizzaArea.content;

Selecting Pizza Size with SwingRadioButton

Listing 5.6 builds the components for pizza size selection. The selection mechanism is 
a SwingRadioButton that sets its toggleGroup  property to pizzaSizeGroup.  This 
enforces the desired mutual exclusion selection behavior (you can only select one size 
at a time). The SwingRadioButton and an associated SwingTextField are grouped into 
an HBox layout component so they appear side by side. Note that the SwingTextField 
component is only enabled when its associated SwingRadioButton is selected. The 
SwingTextField component displays the cost of its associated sized pizza.

The action  event handler sets variable pizzaSizeIndex  to the selected pizza size (0 for 
small, 1 for medium, and 2 for large). This change renders the correct pizza size and 
causes the bound clip  properties in the toppings groups to update their size.

Listing 5.6 Selecting Pizza Size

def pizzaCost = [8.75, 10.75, 12.75];
def pizzaSizeGroup = SwingToggleGroup { }
def smPizza = HBox {
    spacing: 10
    content: [
        SwingTextField {
            columns: 3

  editable: false
            disable: bind not sbSmall.selected

text: " {pizzaCost[0]}"
        }
        sbSmall = SwingRadioButton {
            toggleGroup: pizzaSizeGroup
            text: "Small"

 action: function() {
   pizzaSizeIndex = 0;

            }
        }
    ]
}
def medPizza = HBox {
    spacing: 10
    content: [
        SwingTextField {

            columns: 3
  editable: false

            disable: bind not sbMedium.selected
text: "{pizzaCost[1]}"

        }
        sbMedium = SwingRadioButton {
            toggleGroup: pizzaSizeGroup
            text: "Medium"

 action: function() {
   pizzaSizeIndex = 1;

            }
        }
    ]
}
def lgPizza = HBox {
    spacing: 10
    content: [
        SwingTextField {
            columns: 3

  editable: false
            disable: bind not sbLarge.selected

text: "{pizzaCost[2]}"
        }
        sbLarge = SwingRadioButton {
            toggleGroup: pizzaSizeGroup
            selected: true
            text: "Large"

 action: function() {
   pizzaSizeIndex = 2;

            }
        }
    ]
}

Selecting Toppings with SwingCheckBox

Listing 5.7 builds the object literals for the pizza topping checkboxes. Each checkbox 
sets property text  to the topping name and property selected  to false. The visible

properties of all of the associated graphical •toppingsŽ are bound to the appropriate 
checkbox selected  property (see Listing 5.5 on page 144). When the user selects a top-
ping, the graphical representation of that topping appears on top of the pizza through 
the binding. When the topping is deselected, the graphical representation disappears.

Listing 5.7 Selecting Toppings

def checkPepperoni = SwingCheckBox {
    text: "Pepperoni"
    selected: false
}

